
114. Kaip pasiekti lemputę. Šis uždavinys lengvas, sunkiausia jo vieta – ta, kad prašoma surasti (t. y. spausdinti) tik pirmosios tinkamos dėžės matmenis, o suskaičiuoti reikia visas. Ir visa tai reikia padaryti peržiūrint duomenis vieną kartą (duomenų surašyti į masyvą tikrai neverta – pernelyg paprastas uždavinys, galima taikyti trumpesnį sprendimo būdą).

Sąlygoje pasakyta, kad žmogus gali pasiekti aukštyje h kabančią lemputę. Tarkime, kad žmogus bando pasiekti lemputę nuo dėžės, kurio aukštis yra auk. Užsilipus ant šios dėžės lemputė kabės aukštyje t - auk. Jis galės ją pasiekti, jei t - auk ( h. 

Taigi iš eilės imame visas dėžes ir tikriname. Dėžę bandoma statyti trimis įmanomais būdais. Jei nors vienu atveju žmogus užsilipęs ant dėžės pasieks lubas, dėžė yra tinkama.

Pirmosios tinkamos dėžės numeriui išsaugoti teks panaudoti papildomą kintamąjį: jei spausdintume iš karto, rezultatai displėjaus ekrane susimaišytų su pradiniais duomenimis, – tai nepatogu ir nelogiška.



program kaip_pasiekti_lemputę;

  var n, { dėžių kiekis }

        a, b, c, { dėžės matmenys }

        t, { atstumas nuo grindų iki lemputės }

        h, { žmogus gali pasiekti h aukštyje kabančią lemputę, h <= t }

        nr, { dėžės numeris }

        kiek, { tinkamų dėžių skaičius }

        pirma: integer;{ pirmosios tinkamos dėžės numeris }

begin

  write ('Įveskite dėžių skaičių, kambario aukštį ir');

  writeln (' žmogaus ūgį: ');

  readln (n, t, h);

  kiek := 0;

  pirma := 0;

  for nr := 1 to n do

    begin  { tikrinsime kiekvieną dėžę }

      readln (a, b, c);

      if (a >= t - h) or (b >= t - h) or (c >= t - h)

         then begin   { dėžė tinkama }

                kiek := kiek + 1;

                if pirma = 0   { jei tai pirmoji rasta tinkama dėžė }

                   then pirma := nr

              end

    end;

  writeln (pirma, ' ', kiek)

end.



Testo nr.�Pradiniai duomenys�Rezultatas�Paaiškinimai��1�5 20 15

1 3 2

2 2 2

7 4 1

5 2 6

6 6 3�3 3�Tinka keletas dėžių��2�5 26 10

1 15 14

15 15 15

10 10 10

13 14 15

16 1 1�5 1�Tinka tik viena dėžė��3�4 25 18

1 2 4

2 3 5

6 1 6

4 5 3�0 0�Netinka nė viena dėžė;��

115. Sklypas. Sklypo uždavinys įdomus, tikrai turintis „olimpiadinę ugnelę”, reikalaujantis daugiau matematinio mąstymo ir išprusimo negu algoritmavimo ir programavimo. Reikia gerai pagalvoti ir įsitikinti, kad tveriamos tvoros plotas visuomet gali būti išreiškiamas stačiakampiu, t. y. kad ir kaip „išvingiuotume” pagal nurodytas taisykles duoto ilgio tvorą, ją galima „ištiesinti” į stačiakampį (bus aptvertas tas pats plotas). Tuomet uždavinio algoritmas labai paprastas. Tereikia perrinkti visų galimų stačiakampių plotus.

Iš pradžių plotą galime skaičiuoti ne metrais, o dešimtimis metrų (nes vienos atkarpos ilgis yra 10 m). Žinome, kad 100 a ( 10000 m2, t. y. 1a ( 100 m2. Kvadrato, kurio kiekvienos kraštinės ilgis lygus vienai atkarpai, plotas yra 10 ( 10 m2, t. y. lygus vienam arui. 

Taigi skaičiuodami plotą dešimtimis kvadratinių metrų, jį iš karto skaičiuojame arais.



procedure plotas(n: integer; { atkarpų skaičius }

                 var ha, a: integer); { sklypo dydis hektarais ir arais }

  var plotas, plotis, ilgis: integer;

begin

  plotis := 0;

  plotas := 0;

  repeat

    plotis := plotis + 1;

    ilgis := (n - 2 * plotis) div 2;

    if plotis * ilgis > plotas

       then plotas := plotis * ilgis

  until plotis >= ilgis;

  ha := plotas div 100;

  a := plotas mod 100

end; { plotas }



Jei būtumėte girdėję iš matematikos ar patys sugebėtume įrodyti teiginį, kad to paties perimetro stačiakampių plotas didžiausias tuomet, kai figūra artimiausia kvadratui, tai galėtume dar supaprastinti algoritmą. 

Kaip gauti artimiausią kvadratui stačiakampį? Aišku, kad duotą atkarpų skaičių reikia dalyti iš 4. Ką daryti, jei n tiksliai neišsidalija? Tuomet panagrinėkime liekanas. Jos gali būti tik 1, 2, arba 3. Kai lieka 1 atkarpa, nieko nepadarysi, atkarpos niekur nebegali panaudoti. Kai lieka 2 atkarpos, tuomet galima pridėti po vieną prie dviejų kvadrato kraštinių ir gausime stačiakampį. Kai lieka 3 atkarpos, tai dvi galima panaudoti prailginant kvadrato kraštines kaip ir ankstesniu atveju, o viena liks.

Algoritmas dar supaprastėja:

procedure plotas(n: integer; { atkarpų skaičius }

                 var ha, a: integer { sklypo dydis hektarais ir arais });

  var plotas, plotis, ilgis: integer;

begin

  plotis := n div 4;   { kvadrato kraštinė }

  if (n mod 4 = 2) or (n mod 4 = 3)  { liko 2 arba 3 atkarpos }

     then ilgis := plotis + 1

     else ilgis := plotis;

  plotas := ilgis * plotis;

  ha := plotas div 100;

  a := plotas mod 100

end; { plotas }



Testo nr.�Pradiniai duomenys�Rezultatas�Paaiškinimai��1�4�0ha 1a�Sklypo forma – tikslus kvadratas;��2�100�6ha 25a�Sklypo forma – tikslus kvadratas;��3�101�6ha 25a�n mod 4 ( 1;��4�30�0ha 56a�n mod 4 ( 2;��5�151�14ha 6a�n mod 4 ( 3;��

116. Elektroninis laikrodis. Iš sąlygos suprantame, kad turime du pradinius duomenis – horizontalių ir vertikalių pagaliukų skaičius atskirai.

Uždavinio sąlyga nėra išbaigta. Neaišku, ar laikrodis rodo tik valandas ir minutes, ar ir sekundes. Taip pat neaišku, ar pavyzdžiui, trečią valandą ir penkias minutes laikrodis rodo 03:05, ar 3:05, ar 3:5. Taigi susitarsime, kad laikrodis nerodo sekundžių, ir kad rodydamas valandas, laikrodis neprirašo papildomų nulių, o rodydamas sekundes, nulį prirašo. 

Pirmiausia šiam algoritmui reikia sudaryti procedūrą, kuri nustatytų, kiek horizontalių ir kiek vertikalių pagaliukų yra kiekviename skaitmenyje. Ši procedūra bus reikalinga tiriant elektroninio laikrodžio rodomas valandas ir minutes.

procedure pag (skaitmuo: integer; var hor, ver: integer);

{ randa horizontalių ir vertikalių pagaliukų skaičių skaitmenyje }

begin

  case skaitmuo of

    0: hor := hor + 2;

    2, 3, 5, 6, 8, 9: hor := hor + 3;

    4, 7: hor := hor + 1;

  end;

  case skaitmuo of

    0, 8: ver := ver + 4;

    1, 2, 3, 5, 7: ver := ver + 2;

    4, 6, 9: ver := ver + 3;

  end

end; { pag }



Kadangi valandų ir minučių nėra daug (24valandos paroje ir 60 minučių valandoje), tai paprasčiausia peržvelgti visus galimus laikrodžio rodomų valandų ir minučių variantus. Peržiūrint reikia nustatyti, kiek kiekvienam atvejui sudaryti reikia atskirai horizontalių ir atskirai vertikalių pagaliukų ir iš karto tikrinti, ar laikrodžio rodomas laikas sutampa su duotais pradiniais duomenimis.

Šis uždavinys buvo skirtas jaunesniesiems moksleiviams pirmajame olimpiados etape, todėl stengėmės pateikti paprastesnį, nors ir ne patį efektyviausią laiko atžvilgiu.

Pastaba. Pradiniuose duomenyse pirma pateikiamas horizontalių, po to vertikalių pagaliukų skaičius.



procedure laikas (pagh, pagv: integer; { pagaliukų skaičiai }

                  var v, min: integer);

{ randa laikrodžio rodomą laiką }

  var hor_v, hor_m, ver_v, ver_m: integer;

      rasta: boolean;

begin

  rasta := false;

  v := -1; min := -1; { pradėsime nuo 0 val., 0 min. }

  while (v < 23) and not rasta do

    begin

      v := v + 1;

      { randame pagaliukų skaičių valandoms }

      hor_v := 0; ver_v := 0;

      if v > 10

         then pag (v div 10, hor_v, ver_v);

      pag (v mod 10, hor_v, ver_v);

      while (min < 59) and not rasta do

       begin

         min := min + 1;

         { randame pagaliukų skaičių minutėms }

         hor_m := 0; ver_m := 0;

         pag (min div 10, hor_m, ver_m);

         pag (min mod 10, hor_m, ver_m);

         { tinkamą laiką spausdiname }

         rasta := (hor_v+hor_m = pagh) and (ver_v+ver_m = pagv)

       end;

      if not rasta then min := -1

    end

end; { laikas }



Testo nr.�Pradiniai duomenys�Rezultatai�Paaiškinimai��1�0 6�1 : 11�Paprastas pavyzdys��2�2 6�7 : 17���3�12 10�22 : 28�Galimi tik tokie atsakymai, kur valandų ir minučių skaičius – dviženklis. Kiti galimi atsakymai: 22:38, 22:58, 23: 28, 23:38, 23:58;��4�2 7�1 : 47�Valandų skaičius vienženklis. Kiti galimi atsakymai: 1:47, 4:17, 7:14, 7:41;��

Vyresniems (X – XII kl.)

117. Skaičių kriptograma. Tai labai paprastas uždavinys. Deja, jo sąlyga šiek tiek nekorektiška: gaunama labai daug sprendinių (net 668 948), jie netelpa ekrane, spausdinimas užima beveik minutę. Pasirinkome kitą variantą – sprendinius surašėme į bylą. Ir tokia išeitis nėra labai gera: bylos dydis – apie 17 K. Todėl geriau būtų reikalauti išspausdinti ne visus sprendinius, o tik jų kiekį. Pateikiame algoritmą nepakeitę sąlygos, jūs galite jį nesunkiai modifikuoti.

Sutarsime, kad duotas kriptogramos išdėstymas yra natūralus, įprastas, pavyzdžiui, nėra nereikšminių nulių ar net viso skaičiaus, sudaryto tik iš nulių. Todėl daugiklyje neturi būti nulinių skaitmenų.

Taigi, pirmasis daugiklis gali būti bet koks skaičius iš intervalo [1000, 9999], antrasis – bet koks skaičius iš intervalo [100, 999], kurio užraše nėra nulių. Keturženklį skaičių dauginant iš vienženklio gaunamas skaičius, iš intervalo [1000, 89991], keturženklį dauginant iš triženklio gaunamas skaičius iš intervalo [111000, 9989001]. Todėl būtina patikrinti ar tarpinės sandaugos bei galutinė sandauga turi tiek skaitmenų, kiek yra nurodyta sąlygoje.



procedure kriptograma (var f: text);

  var dauginamasis, daugiklis, sandauga: longint;

      a, b, c: 1..9;    { skaitmenys }

begin

 for dauginamasis := 1000 to 9999 do

  for a := 1 to 9 do

   for b := 1 to 9 do

    for c := 1 to 9 do

     begin

      daugiklis := a * 100 + b * 10 + c;

      sandauga := dauginamasis * daugiklis;

      if (dauginamasis*c <= 9999) and { mažesnis už 1000 }

         (dauginamasis*b >= 10000) and { didesnis už 99999 }

         (dauginamasis*a >= 10000) and { didesnis už 99999 }

         (sandauga <= 9999999)    { mažesnis už 1000000 }

      then writeln(f, dauginamasis, daugiklis:10, sandauga:10)

     end

end; { kriptograma }



Rezultatai�Paaiškinimai��1112 991 1101992���1112 992 1103104�Pateikėme pirmuosius tris sprendinius��1112 993 1104216���...���

118. Dešimtainės trupmenos keitimas paprastąja. Visas uždavinio sunkumas – kaip iš dešimtainės trupmenos užrašo išskirti sveikuosius skaičius: skaitiklį ir vardiklį. Matematiškai tai būtų paprasta: dešimtainę trupmeną daugintume iš dešimties tol, kol gautume sveikąjį skaičių. tai ir būtų skaitiklis. Programavime šitaip daryti nelabai tinka, nes realiųjų skaičių operacijos atliekamos apytiksliai ir nesame garantuoti, kad neperšoksime reikiamos reikšmės (pavyzdžiui, vartodami operaciją trunc ir norėdami nustatyti, ar gavome sveikąjį skaičių).

Todėl šiam uždaviniui būtina sudaryti savą procedūrą dešimtainei trupmenai skaityti. Tuo labiau, kad ši trupmena negali turėti daug skaitmenų (nedaugiau dešimties, nes kitaip negalėsime atlikti veiksmų su paprastąja trupmena). Atkreipiame dėmesį, kad pradiniuose duomenyse trupmeninę dalį nuo sveikosios galime skirti tiek kableliu (pagal lietuvių kalbą), tiek tašku (pagal anglų kalbą).



procedure skaitymas (var skaitiklis, vardiklis: longint;

                     var blogai: boolean);  

{ dešimtainės trupmenos skaitymas }

{ trupmeninę dalį galime skirti tiek kableliu, tiek tašku }

  var c: char;

      kablelis: boolean;

begin

  blogai := false;   { dešimtainė trupmena užrašyta teisingai }

  skaitiklis := 0; vardiklis := 1;

  kablelis := false; { dar nesutiktas kablelis }

  repeat

    read (c);

    if c in ['0'..'9']    { jei skaitmuo }

    then 

     begin

      skaitiklis := skaitiklis * 10 + (ord(c)-ord('0'));

      if kablelis

         then vardiklis := vardiklis * 10

     end

    else if (c = '.') or (c = ',')

         then kablelis := true

         else blogai := true  { blogai užrašyta dešimtainė trupmena }

  until eoln

end; { skaitymas }



Trupmeną prastiname pasinaudodami didžiausio bendrojo daliklio funkcija.

procedure prastinimas (var a, b: longint);

{ trupmenos prastinimas }

  var d: longint;

  function dbd (x, y: longint): longint;

  { didžiausias bendrasis daliklis }

  begin

    if x = 0 then dbd := y

             else dbd := dbd(y mod x, x)

  end;

begin

  d := dbd(a, b);

  a := a div d;

  b := b div d

end; { prastinimas }



Pagrindinės programos dalies nepateikiame: joje paeiliui vieną kartą iškviečiamos abi funkcijos bei išspausdinamas rezultatas.



Testo nr.�Pradinis duomuo�Rezultatas�Komentarai��1�0,3�3/10�Nesuprastinama trupmena��2�0,07�7/100�Sudėtingesnė nesuprastinama trupmena��3�0.5�1/2�Suprastinama trupmena��4�0,375�3/8�Sudėtingesnė suprastinama trupmena��5�5,25�21/4�Trupmena, turinti sveikąją dalį��6�0,000001�1/1000000�Po kablelio daug nereikšminių nulių��7�0.99999999�99999999/100000000�Ilga trupmena��

119. Kubelis su skaičiais. Šis uždavinys lavina sprendžiančiojo vaizduotę. Reikia gerai įsivaizduoti kubelio sieneles ir kaip keičiasi jų padėtys vartaliojant kubelį. Jei nepajėgiame įsivaizduoti, reikia pasidaryti maketą arba bent nusipiešti. Visų keturių komandų pasekmės parodytos 119.1 paveiksle.

119.1 pav.

Matome, kad verčiant kubelį pasikeičia tik keturių sienų padėtys. Tarkime, kad kubelį versime ant kairiojo šono. Tada ant priekinės ir galinės sienelės liks užrašyti tie patys skaičiai. Likusius skaičius užrašome laikrodžio rodyklės apėjimo tvarka, pradedant skaičiumi, užrašytu ant kubelio viršaus. Gausime: a, f, b, e. Kubelį pavertus ant kairiojo šono, ant kubelio viršaus bus užrašytas skaičius  Kubelį pavertus ant kairiojo šono, ant kubelio viršaus bus užrašytas skaičius f. T. y. gausime seką f, b, e, a. Matome, kad verčiant kubelį ant sienelių užrašyti skaičiai pasislenka per vieną.

Sudarome procedūrą, kuri pakeis ant šių sienų užrašytus skaičius (keturių kintamųjų reikšmes). Kreipinys į procedūrą priklauso nuo to, kurią komandą atliekame: reikia atidžiai surašyti sienelių pasikeitimų seką.



program kubelis_su_skaičiais;



  procedure versk (var sien1, sien2, sien3, sien4

                                              : integer);

  { paverčia kubelį }

    var x : integer;

  begin

     x := sien1;        { verčiant }

     sien1 := sien2;    { pasikeičia }

     sien2 := sien3;    { tik keturių }

     sien3 := sien4;    { sienelių padėtys }

     sien4 := x

  end; { versk }



  var a, b, c, d, e, f,  { pradiniai duomenys - skaičiai ant sienelių }

      viršutinė, apatinė,

      priekinė, galinė,

      kairioji, dešinioji: integer;

      r: char;            { pradiniai duomenys – komandų seka }

begin

  readln (a, b, c, d, e, f);   { skaičiai ant sienelių }

  viršutinė := a;   apatinė := b;   priekinė := c;

  galinė := d;      kairioji := e;  dešinioji := f;

  while not eoln do

   begin

    read (r);

    case r of

     'K': versk(viršutinė, dešinioji, apatinė, kairioji);

     'D': versk(viršutinė, kairioji, apatinė, dešinioji);

     'P': versk(galinė, apatinė, priekinė, viršutinė);

     'A': versk(galinė, viršutinė, priekinė, apatinė)

    end

   end;

  writeln (viršutinė)

end.



Testo nr.�Pradiniai duomenys�Rezultatas�Paaiškinimai��1�1 2 3 4 5 6

K�6�ar gerai veikia komanda „kairėn“��2�1 2 3 4 5 6

D�5�ar gerai veikia komanda „dešinėn“��3�1 2 3 4 5 6

P�4�ar gerai veikia komanda „priekin“��4�1 2 3 4 5 6

A�3�ar gerai veikia komanda „atgal“��5�1 2 3 4 5 6

KKK�5�vienodų komandų seka��6�1 2 3 4 5 6

KDKD�1�ar grįžta į pradinę padėtį��7�1 2 3 4 5 6

AAPPDKAPPDA�1�įvairių komandų seka��8�1 2 3 4 5 6

KDPAKDPAKKKAAADDDPPP�5�ilga seka��

II etapas

Jaunesniems (iki IX kl.)

120. Sultis gerti sveika. Pirma reikia patikrinti, kuri savaitės diena – jei pirmadienis, iš karto pasiimti kišenpinigius. O tuomet pirkti sulčių už visus turimus pinigus, parduoti butelius ir taip kartoti veiksmus, kol užteks pinigų (nepamirštant pasiimti kišenpinigių pirmadieniais).

function sultys (k, { kišenpinigių suma }

                 s, { sulčių butelio kaina }

                 b, { tuščio butelio kaina }

                 i, { savaitės diena, kai pradedamos gerti sultys }

                 n: integer { turimų pinigų kiekis }): integer;

  var but: integer;

begin

  but := 0;

  if n = 1          { jei pirmadienis - iš karto paimami kišenpinigiai }

     then n := n + k;

  while n >= s do            { kol užtenka pinigų sulčių buteliui }

    begin

      but := but + n div s;             { perkamos sultys }

      n := n mod s + (n div s) * b; { ir parduodami buteliai }

      if i = 7

         then begin

                i := 1;      { pirmadienį }

                n := n + k   { gaunami kišenpinigiai }

              end

         else i := i + 1

    end;

  sultys := but

end; { sultys }



Testo nr.�Pradiniai duomenys�Rezultatas�Paaiškinimai��1�k = 25, s = 10, b = 3, i = 2, n = 15�1�Pirmąją dieną nusiperkamas tik vienas butelis��2�k = 24, s = 10, b = 2, i = 1, n = 1�2�Prieš perkant sultis pirmąją dieną, būtina pasiimti kišenpinigius��3�k = 25, s = 9, b = 2, i = 4, n = 23�3�Pinigų trečiajam buteliui užteks tik tuomet, jei bus parduoti abu buteliai��4�k = 100, s = 10, b = 2, i = 5, n = 500�74�Didesnis testas, kai buteliai perkami kelias dienas, o tarp dienų yra pirmadienis��

121. Skaitmuo sekoje. Fibonačio seka – klasika tapęs uždavinys. Jį galime rasti kiekviename programavimo vadovėlyje. Čia uždavinio sąlyga šiek tiek pasukta kitaip: reikia rasti skaitmenį sekoje. Tam reikia nuosekliai skaičiuoti, kiek iš viso yra sekos narių skaitmenų. Taip pasieksime reikiamą skaitmenį.

Atkreipkite dėmesį į uždavinio gana patogų ir vaizdų suskaidymą funkcijomis.



function kiek_skaitm (n: integer): integer;

{ kiek skaitmenų turi skaičius n }

  var kiek: integer;

begin

  kiek := 0;

  while n <> 0 do

    begin

        kiek := kiek + 1;

        n := n div 10

    end;

  kiek_skaitm := kiek

end; { kiek_skaitm }



function skaitmuo (n, i: integer): integer;

{ tiriamas sekos narys n, kuriame turi būti ieškomas skaitmuo }

begin

  i := i - 1;

  while i <> 0 do

    begin

      n := n div 10;

      i := i - 1

    end;

  skaitmuo := n mod 10

end; { skaitmuo }



function fib (k: integer): integer;

{ randa k-ąjį Fibonačio skaičių sekos skaitmenį }

  var buvo, { kiek skaitmenų ištirta }

      narys1, narys2, n, { Fibonačio sekos nariai }

      sktm: integer; { Fibonačio sekos nario skaitmenų skaičius }

begin

  buvo := 0;   { kiek skaitmenų buvo ištirta }

  narys1 := 1;  { pirmasis Fibonačio sekos narys }

  narys2 := 1;  { antrasis Fibonačio sekos narys }

  while buvo < k do

    begin

      n := narys1; { imame neištirtą Fibonačio sekos narį }

      sktm := kiek_skaitm(n); { kiek narys turi skaitmenų }

      buvo := buvo + sktm;

      narys1 := narys2;

      narys2 := n + narys2 { randame tolesnį sekos narį }

    end;

  { ieškomas skaitmuo yra nario n  (buvo-k+1)-asis skaitmuo }

  { surandame jį }

  fib := skaitmuo(n, buvo-k+1)

end; { fib }



Testo nr.�Pradiniai duomenys�Rezultatas�Komentaras��1�1�1�Paprastas atvejis��2�19�4�Rezultatas – paskutinis (trečiasis) 12-ojo Fibonačio sekos nario skaitmuo��3�20�2�Rezultatas – pirmasis 13-tojo Fibonačio sekos nario skaitmuo��4�21�3�Rezultatas – antrasis 13-tojo Fibonačio sekos nario skaitmuo��5�36�2�Atsitiktinis testas��6�149�8�Atsitiktinis testas��

122. Spėjami skaitmenys. Tik iš pirmo žvilgsnio šio uždavinio formuluotė atrodo neįprasta. Šiaip uždavinys paprastas. Pirmiausia suskaičiuojame, kiek skirtingų skaitmenų turi spėjamas skaičius. Toliau imame po vieną spėjamą skaitmenį ir tikriname, ar jis įeina į spėjamą skaičių. Jei skaitmuo yra skaičiuje, neatspėtų skaitmenų skaičių mažiname vienetu. Jeigu neliko neatspėtų skaičių, įsimename, kurį spėjimą atliekant skaičius buvo atspėtas. 



program skaičiaus_spėjimas;

  const MAX = 6; { spėjamų skaitmenų sekos ilgis }



  function kiek_skaitm (n: integer): integer;

  { kiek skaičius n turi skirtingų skaitmenų }

    var k: integer; { skirtingų skaitmenų skaičius }

        buvo: set of 0..9;

  begin

    k := 0;  buvo := [];

    while n <> 0 do

      begin

        if not ((n mod 10) in buvo)

           then begin

                  buvo := buvo + [n mod 10];

                  k := k + 1

                end;

        n := n div 10

      end;

    kiek_skaitm := k

  end; { kiek_skaitm }



  function atspėta (n, s: integer): boolean;

  { ar skaitmuo s yra skaičiuje n }

    var yra: boolean;

  begin

    yra := false;

    while not yra and (n <> 0) do

      begin

        yra := n mod 10 = s;

        n := n div 10;

      end;

    atspėta := yra

  end; { atspėta }



  var n,      { sugalvotas skaičius }

      sk,     { spėjamas skaitmuo }

      liko,   { liko neatspėta skaitmenų }

      ats,    { atspėta per tiek bandymų }

      i: integer;

begin

  write ('Įveskite skaičių, kurį reikia atspėti: ');

  readln (n);

  liko := kiek_skaitm (n);

  ats := 0;

  for i := 1 to MAX do  { šeši spėjami skaitmenys }

    begin

      write('Įveskite ', i,'-tąjį spėjamą skaitmenį: ');

      readln(sk);

      if atspėta (n, sk) { jei atspėtas vienas skaitmuo }

         then liko := liko - 1;

      if (liko = 0) and (ats = 0)

         then ats := i { atspėta per i spėjimų }

    end;

  if liko > 0

     then writeln ('NE ', liko)

     else writeln ('TAIP ', ats)

end.



Testo nr.�Pradiniai duomenys�Rezultatas�Komentaras���9

4 8 6 1 0 9�TAIP 6�Skaičius atspėtas, prireikė visų šešių spėjimų; pradinis duomuo – vienženklis;��1�22222

0 4 2 6 1 3�TAIP 3�Skaičius atspėtas; pradinio duomens skaitmenys vienodi;��2�3427

1 6 0 8 5 9�NE 4�Neatspėtas nei vienas skaitmuo;��3�6609

6 4 5 8 9 2�NE 1�Liko neatspėtas tik vienas skaitmuo (0);��4�6641

9 8 4 7 1 5�NE 1�Neatspėti liko du šešetai (nors jie ir du, bet skaitmuo tas pats).��

Vyresniems (X – XII kl.)

123. Abėcėlė. Tai trumpas ir paprastas uždavinio sprendimas, kuris turėtų būti aiškus iš paties algoritmo. Ši programa labai tiktų tiems, kas pradeda mokytis rekursiją ir nori ją geriau perprasti. Rekursinė procedūra skirta visiems galimiems žodžiams formuluoti. 

program abėcėlė;

  const max = 1000;  { maksimalus sudaromų žodžių skaičius }

        MILG = 25;   { maksimalus sudaromų žodžių ilgis }

  type žodis ( array [1..MILG] of char;

  var abc : string; { abėcėlės simboliai }

      žod: žodis;

      f : text;



  procedure sudaryti_žodžius (n, k, poz: integer);

  { rekursinė procedūra, formuojanti žodžius nuo poz pozicijos }

  { abc bei žod – globalieji }

    var i: integer;

  begin

    if poz = k + 1   { jei žodis jau sudarytas }

       then begin { spausdiname jį }

              for i := 1 to k do

                write (f, žod[i]);

              writeln (f);

            end  

       else for i := 1 to n do  { formuojame žodį }

              begin

                žod[poz] := abc[i];

                sudaryti_žodžius (n, k, poz+1, kiek)

              end

  end; { sudaryti_žodžius }



  var n, k, i : integer;

begin

  assign(f, 'ABC.DAT'); 

  reset(f);

  readln(f, n);

  for i: = 1 to n do 

    readln(f, abc[i]);  { skaitoma abėcėlė }

  readln(f, k);

  close(f);

  assign(f, 'ABC.REZ'); 

  rewrite(f);

  žodis(n, k, 1);

  close(f);

end.



Jei pradiniai duomenys būtų gerokai didesni ir visi surasti žodžiai netilptų į rezultatų masyvą, tuomet būtų galima radus žodį iš karto jį rašyti į bylą, o ne į rezultatų masyvą.



Testo nr.�Pradiniai duomenys�Rezultatas�Paaiškinimai��1�1

G

5�GGGGG�Abėcėlę sudaro tik vienas simbolis��2�2

A

Z

6�AAAAAA

AAAAAZ

AAAAZA

...�Abėcėlės simbolių skaičius mažesnis už sudaromų žodžių ilgį��3�4

A

B

C

D

3�AAA

AAB

AAC

AAD

...�Abėcėlės simbolių skaičius didesnis už sudaromų žodžių ilgį��

124. Nuliai skaičiaus gale. Norint išspręsti šį uždavinį, reikia mokėti nustatyti, keliais nuliais baigiasi dešimtainis skaičius n užrašytas sistemoje sist. Skaičius n sistemoje sist užrašomas taip:

n10 ( a0 + a1sist + a2sist2 + a3sist3 + ... + aksistk;

arba

            ----------------- 

n10 ( ak ...a2a1a0sist

Taigi jei skaičių n padalinę iš sistemos pagrindo, gausime 0, ir tai reikš, kad n užrašas sistemoje sist tikrai baigiasi nuliu. Atmetę tą nulį (n :( n div sist) ir toliau kartodami tuos pačius veiksmus rasime, keliais nuliais baigiasi n užrašas.

Beje, sprendinys visuomet egzistuos. n-ainėje sistemoje skaičiaus n užrašas visuomet bus n10 ( 10n

function kiek_nul (n, sist: integer): integer;

{ keliais nuliais baigiasi skaičius n sistemoje sist }

  var kiek: integer;

begin

  kiek := 0;

  while n mod sist = 0 do

    begin

      kiek := kiek + 1;

      n := n div sist

    end;

  kiek_nul := kiek

end; { kiek_nul }



function ilgis (n, sist: integer): integer;

{ randa skaičiaus n skaitmenų kiekį sistemoje sist }

  var ilg, s: integer;

begin

  ilg := 0;

  while n <> 0 do

    begin

      ilg := ilg + 1;

      n := n div sist

    end;

  ilgis := ilg

end; { ilgis }



function sistema (n: integer): integer;

{ randa sistemą, kurioje skaičiaus n užrašas baigiasi }

{ daugiausia nulių }

  var max,  { maksimalus rastų nulių skaičius }

      s,    { tiriama skaičiavimo sistema }

      ss,   { rasta skaičiavimo sistema }

      k: integer;

begin

  max := 0;

  s := 2;      { pradedama nuo dvejetainės sistemos }

  while max < ilgis (n, s) do

    begin

      k := kiek_nul (n, s);

      if k > max

         then begin

                max := k;

                ss := s

              end;

      s := s + 1   { tirsime didesnę skaičiavimo sistemą }

    end;

  if max = 0

     then sistema := n

     else sistema := ss

end; { sistema }



Testo nr.�Pradiniai duomenys�Rezultatas�Komentaras��1�45�3�Vienintelis galimas atsakymas, baigiasi dviem nuliais;��2�55�5�Kiti galimi atsakymai: 11 ir 55; baigiasi vienu nuliu;��3�17�17�Baigiasi vienu nuliu;��4�2450�5�Kiti galimi atsakymai: 7 ir 35; baigiasi vienu nuliu;��5�12005�7�Vienintelis galimas atsakymas; baigiasi 4 nuliais;��

125. Žodynas. Uždavinio sprendimas gana paprastas, jį sudaro šios dalys, išplaukiančios betarpiškai iš uždavinio sąlygos:

a) Lietuviškų raidžių prioritetų nustatymas: aukščiausią prioritetą žymėsime skaičiumi 1, jį turės raidės A ir Ą, žemiausias – raidės Ž prioritetas. Jis lygus 25;

b) Žodžių rikiavimas. Žodžius galima rikiuoti bet kokiu rikiavimo algoritmu. Pasirinkome rikiavimą sukeitimu. Rikiavimo metu teks lyginti du žodžius. 

Žodžių palyginimui teks sudaryti atskirą algoritmą. Imami du gretimi žodžiai ir lyginami paraidžiui, kol randamos nesutampančios raidės arba pasiekiama žodžio pabaiga. Pirmuoju atveju aukščiau bus tas žodis, kurio raidės prioritetas mažesnis, antruoju – tas, kurio ilgis mažesnis.

const MAX = 100;   { maksimalus žodžių skaičius }

      ILG = 20;    { ilgiausias žodis }

type  žodis = string [ILG];

      tžodžiai = array [1..MAX] of žodis;

      prioritetai = array [char] of integer;



procedure rasti_prioritetus (var pr: prioritetai);

{ randa raidžių prioritetus }

begin

  pr['A']:=1;   pr['Ą']:=1;

  pr['B']:=2;   pr['C']:=3;   pr['Č']:=4;

  pr['D']:=5;

  pr['E']:=6;   pr['Ę']:=6;   pr['Ė']:=6;

  pr['F']:=7;   pr['G']:=8;   pr['H']:=9;

  pr['I']:=10;  pr['Į']:=10;  pr['Y']:=10;

  pr['J']:=11;  pr['K']:=12;  pr['L']:=13;

  pr['M']:=14;  pr['N']:=15;  pr['O']:=16;

  pr['P']:=17;  pr['R']:=18;  pr['S']:=19;

  pr['Š']:=20;  pr['T']:=21;

  pr['U']:=22;  pr['Ų']:=22;  pr['Ū']:=22;

  pr['V']:=23;  pr['Z']:=24;  pr['Ž']:=25;

end; { rasti_prioritetus }



function prieš (žod1, žod2: žodis; pr: prioritetai): boolean;

{ patikrina, ar pirmasis žodis turi eiti prieš antrąjį }

  var i, ilgis: integer;

begin

  { rasime trumpesniojo žodžio ilgį }

  if length(žod1) <= length(žod2)

     then ilgis := length(žod1)

     else ilgis := length(žod2);

  i := 1;

  while (pr[žod1[i]] = pr[žod2[i]]) and (i < ILGIS) do

    i := i + 1;

  prieš := (pr[žod1[i]] < pr[žod2[i]]) or

             (pr[žod1[i]] = pr[žod2[i]]) and 

             (length(žod1) < length(žod2));

end; { prieš }



procedure rikiuoti (kiek: integer; { žodžių skaičius }

                    var žodžiai: tžodžiai);

{ žodžiai surikiuojami naudojant Rikiavimą sukeitimu }

  var pr: prioritetai;

      keista: boolean;

      i: integer;

      tarp: žodis;

begin

  { surasime raidžių prioritetus }

  rasti_prioritetus (pr);

  { rikiuosime žodžius }

  keista := true;  { ar du žodžiai sukeisti vietomis }

  while keista do  { pradedama rikiuoti }

    begin

      keista := false;

      for i := 1 to kiek - 1 do

        if prieš (žodžiai[i+1], žodžiai[i], pr)

           { jei (i+1)-as žodis eina prieš i-ąjį }

           then begin   { žodžius keičiame vietomis }

                   keista := true;

                   tarp := žodžiai[i];

                   žodžiai[i] := žodžiai[i + 1];

                   žodžiai[i + 1] := tarp

                 end

    end;

end; { rikiuoti }



Testo nr.�Duotų žodžių skaičius�Paaiškinimai��1�10�Visi žodžiai vienodo ilgio, lietuviškų raidžių žodžiuose nėra��2�13�Žodžiai skirtingo ilgio, lietuviškų raidžių žodžiuose nėra��3�15�Žodžiai skirtingo ilgio, yra atvejis, kai vienas žodis yra kito pradžia, lietuviškų raidžių žodžiuose nėra��5�14�Žodžiai skirtingo ilgio, žodžiuose yra lietuviškų raidžių��6�67�Sudėtingiausias testas; žodžiai skirtingo ilgio ir su lietuviškomis raidėmis, yra atvejis, kai vienas žodis yra kito pradžia (tarp duomenų yra žodžiai: vaikų, vaikus, vaikučiai): tikrinama, ar atitinkamoms skirtingoms raidėms suteiktas vienodas prioritetas, ��7�14�Visų žodžių pirmosios raidės skirtingos: galima rikiuoti tik pagal pirmąją raidę, žodžiuose lietuviškų raidžių yra��8�11�Yra žodžių, prasidedančių tomis pačiomis raidėmis, t. y. žodžiai rikiuojami pagal antrą, trečią ir t. t. raides, žodžiuose yra lietuviškų raidžių��9�14�Yra vienodą prioritetą turinčių raidžių, žodžiuose yra lietuviškų raidžių��

Pastaba. Kai kurie testai vienas kitą tarsi dubliuoja. Žodžius galima rikiuoti įvairiausiais Taigi norėta atkreipti dėmesį į visus uždavinio „kabliukus“.




126. Šešioliktainių skaičių kodavimas
.
 
Sprendimo pavyzdys



A ( 01


B ( 02

C ( 03

D ( 04

E ( 05

F ( 06

0 ( 00


Likę skaitmenys koduojami vienu juos atitinkančiu dešimtiniu skaitmeniu:



1
 ( 
1



2
 ( 
2
 ir t. t.





Pasirinkus kodavimo būdą svarbu nepamiršti užkoduoti ir pirmojo 
skaitmens, vartojamo 
dešimtainių skaitmenų poroje: šiuo atveju 0. Pamiršus tą padaryti nebūtų galima vienareikšmiškai dekoduoti, pavyzdžiui, tokio užrašo 205: ar 205, ar 2E.


Dar vienas 
pastebėjimas. 
K
adangi nuliai skaičiaus pradžioje nerašomi, tai esant vienodai šešioliktainių skaitmenų tikimybei, nulių bus mažiau, 
todėl pirmuoju kodo poros skaitmeniu tikslinga parinkti nulį.





III etapas

I dalis

Jaunesniems (iki IX kl.)

127. Siųstuvai ir imtuvas. Pradžioje tarkime, kad turime tik vieną siųstuvą ir imtuvą, pirmasis signalas išsiunčiamas laiko momentu 0. Siųstuvas skleidžia signalus kas T1 sekundžių, o šie imtuvą pasiekia po L1 sekundžių. Pirmasis signalas siųstuvą pasieks po L1 sekundžių, o po to kas T1 sekundžių imtuvą pasieks nauji signalai. 

T. y. signalai imtuvą pasieks laiko momentais L1 + mT1, kur m = 0, 1, ....

Pavyzdžiui, L1 = 5, T1 = 2:

Laiko momentai�0�1�2�3�4�5�6�7�8�9�10�11�12�13�...��Siųstuvas�1��2��3��4��5��6��7��8��Imtuvas������1��2��3��4��5���1 lentelė

Įjunkime dar vieną siųstuvą. Sakykime, kad L2 = 2, T2 = 3. Dabar pirmojo siųstuvo signalus žymėsime pliusais, antrojo – minusais.

Laiko momentai�0�1�2�3�4�5�6�7�8�9�10�11�12�13�14�15�16�17�18�19�...��Pirmas siųstuvas�+��+��+��+��+��+��+��+��+��+��+��Antras siųstuvas�–���–���–���–���–���–���–����Imtuvas���–���+,

–��+�–�+��+,

–��+�–�+��+,

–��+�–��2 lentelė

Matome, kad nuo penkto laiko momento gauname vienodus periodus (jie lentelėje pažymėti kita spalva). Periodo ilgis – 6 laiko vienetai, kiekviename periode yra po vieną blogą signalą, taigi imtuvą pasieks 16.7% blogų signalų. 

Bendru atveju blogi signalai gali ateiti dar neprasidėjus pirmajam periodui (šiame pavyzdyje laiko momentais 1, 2, 3, 4). Tačiau blogų signalų skaičius iki prasidedant pirmajam periodui yra fiksuotas (baigtinis), o pačių periodų (ir blogų signalų juose) skaičius – begalinis, nes siųstuvai veikia be perstojo. Todėl skaičiuojant blogų signalų procentą reikia nekreipti dėmesio į signalus pasiekiančius imtuvą ligi pirmojo periodo.

Periodiškai pasikartojančius signalus gavome panagrinėję vieną pavyzdį. Įrodysime kad tai teisinga bendru atveju.

Pradžioje sakykime, kad turime du siųstuvus ir laikykime kad L1 ( L2 ( L. Tuomet pirmasis abiejų siųstuvų signalas (blogas) imtuvą pasieks laiko momentu L. Antrasis blogas signalas pasieks imtuvą laiko momentu L + mbk (T1, T2), čia mbk – mažiausias bendrasis kartotinis, t. y. kai galios lygybė m1T1 ( m2T2. Trečiasis blogas signalas pasieks L + 2mbk(T1, T2) ir t. t. Periodas prasidės laiko momentu L ir jo ilgis bus mbk(T1, T2).

Imkime aukščiau pateiktą pavyzdį ir sakykime, kad L1 ( L2 ( 2, T1 ( 2, T2 ( 3.

Laiko momentai�0�1�2�3�4�5�6�7�8�9�10�11�12�13�14�15�16�17�18�19�...��Pirmas siųstuvas�+��+��+��+��+��+��+��+��+��+��+��Antras siųstuvas�–���–���–���–���–���–���–����Imtuvas���+

–��+�–�+��+

–��+�–�+��+

–��+�–�+����3 lentelė

Analogiška situacija bus ir tada kai turėsime n siųstuvų. Laiko momentu L imtuvą pasieks visų siųstuvų pirmasis signalas. Blogi signalai pasieks imtuvą laiko momentais m(mbk(Ti, Tj), čia m ( 0, 1, 2... o i, j – bet kurių dviejų siųstuvų numeriai. Laiko momentu mbk (T1, T2, ... Tn) imtuvą pasieks visų n siųstuvų signalai. Taigi periodas prasidės laiko momentu L ir jo ilgis bus mbk (T1, T2, ... Tn). 

Jei turėtume tik du imtuvus, blogų signalų ieškotume skaičiuodami mbk. Tačiau turint n signalų, taip ieškoti sunku, nes gali sutapti ne tik dviejų bet trijų ar daugiau skaičių mbk.

Pereikime prie atvejo su dviem siųstuvais, kai L1 ( L2. Tarkime, kad L1 ( L2. Tuomet periodas prasidės tuo laiko momentu L1 (t. .y kai abiejų siųstuvų signalai jau bus pasiekę imtuvą po vieną kartą). Mat nuo to momento imtuvą pasiekiantys signalai nebepriklausys nuo Li, tik nuo Ti. Aukščiau pateiktame pavyzdyje (2 lentelė) laiko momentu L1 imtuvą pasiekia ne tik pirmojo, bet ir antrojo siųstuvo signalas. Todėl situacija bus lygiai tokia pati kaip ir tada, kai L1 ( L2. Panagrinėkime kitokį pavyzdį. Sakykime, kad L1 = 3, T1 = 2, L2 = 1, T2 = 3

Laiko momentai�0�1�2�3�4�5�6�7�8�9�10�11�12�13�14�15�16�17�18�19�...��Pirmas siųstuvas�+��+��+��+��+��+��+��+��+��+��+��Antras siųstuvas�–���–���–���–���–���–���–����Imtuvas��–��+�–�+��+

–��+�–�+��+

–��+�–�+��+

–���4 lentelė

Periodas prasidės laiko momentu 3, jo ilgis – 6. 

Analogiškai, kai turime n siųstuvų, periodas prasidės tuo laiko momentu, kai imtuvą pasieks pirmasis siųstuvo, kurio signalai eina ilgiausiai (didžiausias L) signalas. Periodo ilgis bus mbk (T1, T2, ... Tn). 

Sprendžiant uždavinį patogiausia modeliuoti kiekvieną laiko momentą iš pirmojo periodo, t. .y iš intervalo [Lmax, Lmax + mbk (T1, T2, ... Tn)[. 

Na, o blogų signalų procentas lygus:

blogų_signalų_skaičius ( 100 / gautų_signalų_skaičius_viename_periode

const MAX_N = 20; { maksimalus siųstuvų skaičius }

type lentelė = array [1..MAX_N] of integer;



function dbd (x, y: integer): integer;

{ didžiausias bendras daliklis }

begin

  if x = 0 then dbd := y

           else dbd := dbd (y mod x, x)

end; { dbd }



function bmk (x, y: integer): integer;

{ mažiausias bendrasis kartotinis }

begin

  bmk := x div dbd (x, y) * y

end; { bmk }



function blogi_signalai (n: integer; t, l: lentelė)

                                               : integer;

{ apskaičiuoja blogų signalų, kuriuos gauna imtuvas, procentą }

  var blogi, { blogų signalų skaičius }

      viso, sign , l_max, kart, i, j: integer;

      pas: lentelė; { lentelė, rodanti kuriais laiko momentais }

                  { atitinkamo siųstuvo signalas pasiekia imtuvą }

begin

  { rasime l_max }

  l_max := 0;

  for i := 1 to n do

    if l_max < l[i]

       then l_max := l[i];

  { rasime bmk(T1, T2, ... Tn) }

  kart := 1;

  for i := 1 to n do

    kart := bmk (kart, t[i]);

  { rasime kada pirmasis kiekvieno siųstuvo signalas pasiekia imtuvą }

  pas := l;

  { rasime kada kiekvieno siųstuvo pirmasis signalas siekia imtuvą }

  { nagrinėjamame periode }

  for i := 1 to n do

    while pas[i] < l_max do

      pas[i] := pas[i] +  t[i];

  { modeliuosime kiekvieną laiko momentą pirmajame periode }

  blogi := 0;

  viso := 0;

  for i := l_max to l_max + (kart-1) do

    begin

      sign := 0; { momentu i imtuvą pasiekiančių signalų skaičius }

      for j := 1 to n do

        if pas[j] = i

           then sign := sign + 1;

      if sign > 1

         then blogi := blogi + 1;

      if sign > 0

         then viso := viso + 1;

      { randame tolesnius laiko momentus kada signalai pasieks imtuvą }

      for j := 1 to n do

        if pas[j] <= i

           then pas[j] := pas[j] + t[j]

    end;

  blogi_signalai := round (blogi/viso*100)

  { tokia veiksmų tvarka nesukels perpildymo }

end; { blogi_signalai }



Testo nr.�Rezultatas�Komentarai��1�100�Visi signalai blogi (sutampa dviejų siųstuvų charakteristikos)��2�73�Periodo ilgis lygus visų Tk sandaugai��3�40���4�47�Keletas įvairesnių atvejų, kai Tk ir Lk skirtingi��5�73���6�38�Sutampa visų imtuvų L (t. y. neateina blogų signalų iki pirmojo periodo pradžios), nedidelis testas��7�93�Sutampa visų imtuvų L, (t. y. neateina blogų signalų iki pirmojo periodo pradžios)didesnis testas��8�0�Sutampa visų imtuvų T, Visi signalai geri��

128. Pokeris. Šis uždavinys yra 140 uždavinio palengvintas variantas, todėl jam tiktų to uždavinio sprendimo idėja: tereikia ją supaprastinti. Tačiau tame sprendime yra nemažai kombinatorikos elementų, kurių gali nežinoti jaunesni moksleiviai. Todėl pateikiame paprastesnį algoritmą.

Bandysime keisti visas kortas iš eilės. 

Imame pirmąją kortą ir dedame atgal į kaladę. Dabar kaladėje yra dvidešimt kortų. Traukiame iš kaladės kortą ir dedame vietoj pakeistosios. Turime penkių kortų kombinaciją. Reikia patikrinti, ar ši kombinacija yra išlošiama ir jei taip – nustatome kuri iš aštuonių išlošiamų kombinacijų bei padidiname šios kombinacijos variantų skaičių. Taip vietoj trūkstamos dedame visas kaladėje likusias kortas. Išbandę visas kortas, gauname, tokią lentelę (ji atitinka pavyzdį sąlygoje):

Keičiamos kortos numeris – 1��Išlošiamos kombinacijos nr.�1�2�3�4�5�6�7�8��Keliais būdais galima sudaryti tą kombinaciją�12�6�2�0�0�0�0�0��

Iš lentelės matome, kad pakeitus pirmąją kortą didžiausia tikimybė yra sudaryti Porą. Ši tikimybė lygi 12/20 ( 0.6. Atkreipkite dėmesį, kad lentelėje tikimybės neskaičiuojamos. Mat tikimybė sudaryti kažkurią išlošiamą kombinaciją bus lygi variantų skaičiui, padalintam iš dvidešimties. Taigi, kuo didesnis variantų skaičius, tuo didesnė ir tikimybė. 

Užpildytoji lentelę parodo išlošimo galimybes pakeitus pirmąją kortą. Tokia pat lentelė sudaroma dar keturis kartus – keičiant antrąją, trečiąją ir t. t. kortas. Ieškomą tikimybę bus galima apskaičiuoti iš visų penkių lentelių išrinkus didžiausią variantų skaičių ir jį padalinus iš 20.

const K_SK = 5; { duotų kortų skaičius }

      IŠL = 8;  { išlošiamų kombinacijų skaičius }

      GAL_KOMB = 20; { keičiamą kortą galime ištraukti tiek būdų }

type rangas = (dev, deš, val, dam, kar, tūz);

     rūšis = (čir, būg, kry, vyn);

     korta = record

               rū: rūšis;  { kortos rūšis }

               ra: rangas { kortos rangas }

             end;

     kortos = array [1..K_SK] of korta;

     išlošimai = array [1..IŠL] of integer;



function buvo (kr: kortos; { turimos kortos }

               i: integer; { keičiamos kortos numeris }

               k: korta{ naujai įdedamos kortos numeris}): boolean;

{ tikrina, ar nauja korta nesutampa su kuria nors nekeičiama korta }

  var j: integer;

begin

  buvo := false;

  for j := 1 to K_SK do

    if (i <> j) and 

       (k.rū = kr[j].rū) and (k.ra = kr[j].ra)

       then buvo := true

end; { buvo }



procedure išlošimas (kr: kortos;

                     var nr,  { keičiamos kortos numeris }

                         komb: integer; { išlošiama kombinacija }

                     var tik: real);

  var kiek: išlošimai;

      { kelios kombinacijos sudarys i-tąją išlošiamą situaciją }

      keič, k: korta;

      kom_sk, { didžiausias kombinacijų skaičius }

      i, j: integer;

begin

  kom_sk := 0; { nesudarėme nei vienos išlošiamos kombinacijos }

  for i := 1 to K_SK do

    begin { keisime i-tąją kortą }

      for j := 1 to IŠL do

        kiek[j] := 0;

        keič := kr[i]; { įsimename keičiamą kortą }

        { vietoj keičiamos kortos įstatome visas likusias kortas }

        for k.rū := čir to vyn do

          for k.ra := dev to tūz do

            if not buvo(kr, i, k)

               { jei pakeista korta nedubliuoja kurios nors iš paliktų }

               then begin

                      kr[i] := k;

                      laimi (kr, kiek);

                    end;

        { patikrinsime, gal radome išlošiamą situaciją, kurią sudaryti yra }

        { dar didesnė tikimybė }

        for j := 1 to IŠL do

          if kiek[j] > kom_sk

             then begin

                    kom_sk := kiek[j];

                    nr := i;

                    komb := j;

                  end;

        kr[i] := keič; { atstatome pakeistą kortą }

    end;

  { beliko apskaičiuoti tikimybę }

  tik := kom_sk/GAl_KOMB;

end; { išlošimas }



Žemiau pateikta trūkstama procedūra laimi, kuri pagal turimas penkias kortas nustato ar jos sudaro kurią nors išlošiamą kombinaciją ir jei taip, tai kokią. Procedūra gan aiški, todėl jos daug nekomentuosime. Svarbiausia – suskaičiuoti kiek yra kiekvieno rango ir kiekvienos rūšies kortų. Jei pavyko sudaryti aukštesnio prioriteto kombinaciją (pavyzdžiui, Ketvertą), žemesnio prioriteto išlošiamos kombinacijos (pavyzdžiui Trejetas) nebenagrinėjamos.

procedure laimi (kr: kortos; var kiek: išlošimai);

{ tikrina, kurią laimimą situaciją sudaro turima penkių kortų kombinacija }

  var i: rangas;

      j: rūšis;

      ran: array [dev..tūz] of integer;

      rūš: array [čir..vyn] of integer;

      galima: array [1..IŠL] of boolean;

      { ar galima sudaryti i-tąją išlošiamą kombinaciją }

      t: integer;

begin

  for i := dev to tūz do

    ran[i] := 0;

  for j := čir to vyn do

    rūš[j] := 0;

  { suskaičiuosime kiek kokių kortų turime }

  for t := 1 to K_SK do

    begin

      ran[kr[t].ra] := ran[kr[t].ra] + 1;

      rūš[kr[t].rū] := rūš[kr[t].rū] + 1;

    end;

  for t := 1 to IŠL do

    galima[t] := false;

  { tikrinsime pirmąsias keturias kombinacijas }

  for i := dev to tūz do

    if ran[i] = 2

       then if not galima[1]

               then galima[1] := true { Pora }

               else galima[2] := true { Dvi poros }

       else if ran[i] = 3

               then galima[3] := true { Trejetas }

               else if ran[i] = 4

                       then galima[4] := true; { Ketvertas }

  { tikrinsime kombinaciją Pilni namai }

  galima[5] := galima[3] and galima[1];

  { tikrinsime kombinaciją Visos kortos vienos rūšies }

  galima[6] := false;

  for j := čir to vyn do

    if rūš[j] = K_SK

       then galima[6] := true;

  { tikrinsime kombinaciją Visos kortos iš eilės }

  galima[7] := true;

  for i := dev to tūz do

    if (ran[i]>1) or (ran[i]=0) and not (i in [dev,tūz])

       then galima[7] := false;

  { tikrinsime kombinaciją Pokeris }

  galima[8] := galima[6] and galima[7];

  { įsiminsime didžiausią sudarytą kombinaciją }

  t := IŠL;

  while (t > 0) and not galima[t] do

    t := t - 1;

  if t > 0

     then kiek[t] := kiek[t] + 1;

end; { laimi }





Testo nr.�Pradiniai duomenys (išvardinti vienoje eilutėje)�Visi galimi atsakymai�Komentarai��1�K1 K2 V3 B1 B4�K2  V3  B4  K1  B1

1   1   1   1   1

0.8 0.8 0.8 0.8 0.8�Galima keisti bet kurią kortą ir visais atvejais didžiausia tikimybė sudaryti Porą��2�K3 K1 V4 B1 B3�V4

2

0.8�Didžiausia tikimybė sudaryti Dvi poras��3�V3 K3 V4 B1 B3�V4   B1

3    3

0.8  0.8�Didžiausia tikimybė sudaryti Trejetą��4�K4 V4 V3 B4 Č4�V3

4

1�Tikimybė sudaryti Ketvertą lygi 1��5�V2 K3 B2 B3 K2�V2  B2  K2  K3  B3

2   2   2   3   3

0.8 0.8 0.8 0.8 0.8�Galima sudaryti arba Dvi poras arba Trejetą��

Vyresniems (X – XII kl.)

129. Siųstuvai ir imtuvas. Tai buvo vienas sunkiausių Lietuvos informatikos olimpiadų uždavinių. Jo pilnai nepavyko išspręsti nei vienam dalyviui. Sprendimo idėja gana paprasta – perrinkimas. Tiesa, gan sunku jį optimizuoti taip, kad programos vykdymo laikas neviršytų leistino.

Pirmiausia sugeneruojamas teisingas, tačiau nebūtinai optimalus sprendinys. Jis gaunamas naują siųstuvą jungiant tada, kai ankstesnis jau būna pabaigęs darbą. Įsimenama, kada šiuo atveju paskutinysis paskutinio siųstuvo signalas pasiekia imtuvą. Šį momentą pavadinkime pask.

Tuomet perrinkinėjama rekursiškai. Vienu rekursijos žingsniu įjungiamas vienas siųstuvas. Siųstuvą bandoma jungti visais laiko momentais pradedant nuliniu ir baigiant momentu pask. Atkreipkite dėmesį, kad būtina išbandyti visus jungimo variantus. Negalima sustoti suradus vieną tinkamą. Šykštus algoritmas šiam uždaviniui netinka. Kai įjungiami visi siųstuvai, patikrinama, ar rastas optimalesnis sprendinys. Jei taip, jis įsimenamas, grįžtama per žingsnį atgal ir dirbama toliau.

Sprendimas buvo optimizuotas taip:

a) siųstuvai buvo surikiuoti jų pranešimų perdavimo trukmės mažėjimo tvarka. Yra didesnė tikimybė greičiau rasti sprendinį, jei ilgiau dirbančius siųstuvus bandome jungti pirmiau;

b) prieš pradedant perrinkimą kiekvienam siųstuvui buvo apskaičiuota jo skleistinė, t. y. suskaičiuotas laikas, kada kiekvienas siųstuvo signalas pasieks imtuvą, jei siųstuvas buvo įjungtas nuliniu laiko momentu. Taip pat sudaroma lentelė, kurioje pažymėta, kokiais momentais imtuvą pasiekia signalai. Šios lentelės pagreitina darbą, tikrinant, ar kažkokiu momentu galima įjungti siųstuvą, t. y. ar tuo momentu įjungus siųstuvą, imtuvas negaus blogų signalų. 

c) vartojama nemažai globalių kintamųjų. Masyvų, kaip parametrų-reikšmių perdavimas procedūroms ar funkcijoms labai lėtina darbą;

d) Siųstuvą bandoma jungti ne iki momento pask. Žiūrima, ar siųstuvas perduos visą savo pranešimą iki šio momento.

e) Jei paskutinįjį siųstuvą pavyko įjungti, darbas iš karto baigiamas.

const MAX_N    = 20; { maksimalus siųstuvų skaičius }

      LAIK     = 200; { pranešimo perdavimo laikas }

type siųstuv   = array [1..MAX_N] of integer;

     siųstuv_l = array [1..MAX_N] of boolean;

     lentelė   = array [1..MAX_N, 1..LAIK] of integer;

     imtuvas   = array [1..LAIK*MAX_N] of boolean;



{ žemiau surašyti globalieji kintamieji }

var n: integer; { siųstuvų skaičius }

    t, l, p: siųstuv; { siųstuvų charakteristikos }



procedure skleistinė (var skl: lentelė);

{ kada imtuvą pasiekia kiekvieno siųstuvo signalai }

{ jei siųstuvas įjungiamas nuliniu laiko momentu }

  var k, sek, sign: integer;

begin

  for k := 1 to n do

    for sign := 1 to LAIK do

      skl[k, sign] := 0;

  for k := 1 to n do

    begin

      skl[k, 1] := l[k];

      sek := l[k];

      for sign := 2 to p[k] do

        begin

          sek := sek + t[k];

          skl[k, sign] := sek;

       end

    end;

end; { skleistinė }



procedure rikiuoti (ilg: siųstuv; var tvarka: siųstuv);

{ rikiuoja siųstuvus pranešimų perdavimo laiko mažėjimo tvarka }

{ rikiuojama burbuliuko metodu }

  var i, tarp: integer;

begin

  i := 1;

  while i < n do

    if ilg[tvarka[i]] >= ilg[tvarka[i+1]]

       then i := i + 1

       else begin  { sukeičiame i-tąjį ir (i+1)-mąjį elementą }

              tarp := tvarka[i];

              tvarka[i] := tvarka[i + 1];

              tvarka[i + 1] := tarp;

              if i > 1 then i := i - 1

                       else i := i + 1;

            end

end; { rikiuoti }



procedure jungimas (var kada: siųstuv);

{ randa optimalią siųstuvų įjungimo tvarką }

  var skl: lentelė;

      laikas, { kada pradeda veikti kiekvienas siųstuvas }

      tvarka, { kokia tvarka bandysime jungti siųstuvus }

      ilg: siųstuv; { siųstuvų pranešimų perdavimo ilgiai }

      i, k, pirmas, num, sek,

      pask, { laiko momentas, kada imtuvą pasiekia }

            { paskutinis signalas optimaliame sprendinyje }

      pask_e: integer; { ir nagrinėjamame sprendinyje }

      imt: imtuvas;  { kuriais momentais imtuvą pasiekia signalai }



  procedure įjungti (k, sek: integer;

                   var imt: imtuvas; var galima: boolean);

  { ar galima siųstuvą k įjungti laiko momentu sek ? }

  { jei galima, tai pakeičiamas masyvas imt }

    var sign, i: integer;

  begin

    sign := 1;

    gerai := true;

    while (skl[k, sign] <> 0) and gerai do

      begin

        if imt[sek + skl[k, sign]]  { jei vieta užimta }

           then begin { atstatome atgal }

                  for i := sign-1 downto 1 do

                    imt[sek + skl[k, i]] := false;

                  gerai := false

                end

             else imt[sek + skl[k, sign]] := true;

        sign := sign + 1

      end;

  end; { įjungti }



  procedure išjungti (k, { siųstuvo numeris }

                      sek: integer; { kada jis buvo paleistas }

                      var imt: imtuvas);

  { išjungia siųstuvą (atstato masyvą imt) kai grįžtama per žingsnį atgal }

    var sign: integer;

  begin

    sign := 1;

    while skl[k, sign] <> 0 do

      begin

        imt[sek + skl[k, sign]] := false;

        sign := sign + 1

      end;

  end; { išjungti }



  procedure dėlioti (nr, pask_e: integer; 

                     laikas: siųstuv;

                     var kada: siųstuv; 

                     var pask: integer);

  { rekursinė procedūra, dėliojanti siųstuvus }

  { nr – nagrinėjamas siųstuvas }

  { pask_e – kada imtuvą pasiekia įjungtų siųstuvų paskutinis signalas }

    var k, senas, sek: integer;

        galima: boolean;

  begin

    if nr > n { jei visi siųstuvai įjungti }

    then begin      { tai radome geresnį sprendinį }

           if pask_e < pask then

             begin

               kada := laikas;

               pask := pask_e;

             end;

         end

    else begin

           k := tvarka[nr]; { jungsime siųstuvą k }

           sek := 0; { bandysime jungti nulinę sekundę }

           while sek + ilg[k] < pask do

             begin

               įjungti (k, sek, imt, galima);

               { jei siųstuvą galima įjungti sekundę sek }

               if galima then

                  begin

                   senas := pask_e;

                   if pask_e < sek + ilg[k]

                      then pask_e := sek + ilg[k];

                   laikas[k] := sek;

                   dėlioti(nr+1,pask_e,laikas,kada,pask);

                   pask_e := senas;

                   išjungti (k, sek, imt);

                   if nr = n then sek := pask;

                  end;

               sek := sek + 1;

             end;

         end;

  end; { dėlioti }



begin

  skleistinė (skl);

  { randame kiekvieno pranešimo ilgį; bei teisingą, tačiau nebūtinai }

  { optimalų sprendinį }

  for k := 1 to n do

    begin

      tvarka[k] := k;

      ilg[k] := l[k] + (p[k]-1)*t[k];

      if k = 1

         then kada[k] := 0

         else kada[k] := pask + 1;

      pask :=  kada[k] + ilg[k];

   end;

  { siųstuvus surikiuosime pranešimų perdavimo laiko mažėjimo tvarka }

  { tokia tvarka juos vėliau bandysime jungti }

  rikiuoti (ilg, tvarka);

  { parenkame laiką, kada įjungsime ilgiausiai dirbantį siųstuvą }

  pirmas := 0; { ilgiausiai dirbantį įjungiame nuliniu momentu }

  num := tvarka[1];

  while (pirmas + ilg[num]) < pask do { jei dar verta skaičiuoti }

    begin

      for k := 1 to n do  { visi siųstuvai išjungti }

        laikas[k] := 0;

      for i := 1 to pask do { imtuvo nepasiekė nei vienas signalas }

        imt[i] := false;

      { įjungiame ilgiausiai dirbantį imtuvą }

      laikas[num] := pirmas;

      sek := laikas[num] + l[num];

      for i := 1 to p[num] do

        begin

          imt[sek] := true;

          sek := sek + t[num]

        end;

      pask_e := pirmas + ilg[num];

      dėlioti (2, pask_e, laikas, kada, pask);

      pirmas := pirmas + 1; { bandysime jungti sekunde vėliau }

    end;

end; { jungimas }



Testo nr.�Laiko momentas, kada paskutinis signalas pasiekia imtuvą�Paaiškinimai��1�10�Duoti tik trys siųstuvai. Paprastas testas, kurio pagalba lengva ieškoti klaidų sprendime��2�13�Duoti keturi siųstuvai. Jų visos charakteristikos sutampa; šis testas skirtas patikrinti, ar programa sugeba rasti korektišką (tegu ir ne optimalų) sprendinį��3�54�Duoti keturi siųstuvai; pirmiausia jungiamas antras siųstuvas; likusieji trys įjungiami šiam nebaigus darbo��4�19�Duoti trys siųstuvai; jie visi jungiami eilės tvarka��5�269�Duoti šeši siųstuvai; trumpiausias testas, tikrinantis sprendimo efektyvumą��6�70�Duoti septyni siųstuvai; trys siųstuvai (2, 3, 5) turi būti įjungti nuliniu laiko momentu; ilgesnis testas sprendimo efektyvumui tikrinti��7�441�Duoti trys siųstuvai, du iš jų (2, 3) įjungiami nuliniu laiko momentu; paprastas testas tikrinantis, ar sprendime numatytas atvejis, kai keli siųstuvai įjungiami tuo pačiu momentu��8�61�Duota 15 siųstuvų�Šie trys testai taip pat��9�126�Duota 14 siųstuvų�tikrina sprendimo ��10�52�Duota 10 siųstuvų�efektyvumą��

130. Trys vienodi skaitmenys. Pats paprasčiausias sprendimo būdas perrinkimas. Imame pirmąjį skaičių iš intervalo [n, m]. Jį perrašome visomis skaičiavimo sistemomis nuo 2 iki n (t. y. iki paties skaičiaus) ir žiūrime, ar radome dvi sistemas, kuriose skaičius užsirašo trimis vienodais skaitmenimis. Po to imame kitą skaičių iš intervalo ir t. t., kol peržiūrime visus skaičius nuo n iki m.

Norint parašyti tokį algoritmą, reikia žinoti, kaip dešimtainį skaičių s užrašyti p-aine sistema. Skaičių s dalijame iš sistemos pagrindo, tol kol liekana tampa mažesnė už pagrindą, o tuomet gautus skaitmenis perrašome atbulai.

130.1 pav.

Mūsų atveju nėra būtina gauti patį skaičių. Tereikia paimti vieną skaitmenį (paskutinysis skaitmuo bus lygus s mod p) ir patikrinti, ar likusieji skaitmenys jam lygus ir ar skaitmenų skaičius lygus trims.

type vienodi = array [1..2] of  { dviems rastiems skaičiams saugoti }

                 record { pagrindas ir skaitmuo }

                   pagr, sk: integer

                 end;



procedure išvesti (s: integer; v: vienodi; var f: text);

{ išveda į bylą rastą skaičių }

begin

  writeln(f, s);

  writeln(f,v[1].pagr,' ',v[1].sk,' ',v[1].sk,' ',v[1].sk);

  writeln(f,v[2].pagr,' ',v[2].sk,' ',v[2].sk,' ',v[2].sk);

end; { išvesti }



procedure ieškoti_vienodų (m, n: integer; var f: text);

{ randa ieškomus skaičius iš intervalo [n, m] ir juos įrašo į bylą f }

  var s, tiriu, kiek, skaitmuo, p, dvi: integer;

      v: vienodi;

begin

  rewrite (f);

  for s := m to n do  { tirsime visus skaičius iš duoto intervalo }

    begin

      p := 2;   { pradedama nuo mažiausio sistemos pagrindo }

      dvi := 0; { kol kas nerasta nė vienos sistemos }

      repeat  { tiriamą skaičių užrašysime įvairiomis sistemomis }

        tiriu := s;

        skaitmuo := tiriu mod p;

        kiek := 1; { keliais skaitmenimis sistemoje p užsirašo skaičius s }

        tiriu := tiriu div p;

        while (tiriu <> 0) and     { ar visi trys skaitmenys vienodi }

              (kiek < 3) and (tiriu mod p = skaitmuo) do

          begin

            kiek := kiek + 1;

            tiriu := tiriu div p

          end;

        if (kiek=3) and (tiriu=0) { jei trys skaitmenys vienodi }

           then begin

                  dvi := dvi + 1; { jau rastas užrašas viena sistema }

                  v[dvi].pagr := p;

                  v[dvi].sk := skaitmuo

                end;

         p := p + 1

         { tiriame tol, kol randame dvi sistemas }

      until (dvi = 2) or (p > s);

      if dvi = 2 then išvesti (s, v, f)

    end;

  close (f);

end; { ieškoti_vienodų }



Galime pasiūlyti ir kitą, matematiškesnį, tačiau žymiai greičiau veikiantį, sprendimą. 

Tarkime, kad turime skaičiavimo sistemą su pagrindu p. Nesunku pastebėti, kad 

111p = (1 + p + p2)10.                                                                                    (1)

Bet kuris skaičius, kurį p–ainėje skaičiavimo sistemoje galima užrašyti trimis vienodais skaitmenimis (pažymėkime juos sk) dešimtainėje sistemoje užsirašys taip:

sk ( (1 + p + p2), kur 1 ( sk < p.                                                                   (2)

Turime dešimtainį skaičių s. Jis p-ainėje sistemoje užsirašo trimis vienodais skaitmenimis, jei 

0 < s div (1 + p + p2) < p { p-ainės sistemos didžiausias skaitmuo neviršys p }

ir 

s mod (1 + p + p2) = 0     {  iš (2) }

Taigi, imame visas sistemas nuo s iki 2 ir tiriame, ar tenkinamos šios abi sąlygos. Jei randame dvi tinkamas skaičiavimo sistemas: turime vieną iš ieškomų skaičių.

Patogu susidaryti lentelę, kurioje būtų surašyti skaičiaus 111, užrašyto visomis sistemomis nuo 2 iki 10000 (tokia yra maksimali n reikšmė), reikšmės dešimtainėje skaičiavimo sistemoje.

Sistema �2�3�4�5�6�7�…��111n = 1 + p + p2�7�13�21�31�43�57�…��

Aukščiau aprašytas algoritmas yra teisingas, tačiau paėmus didesnį pradinių duomenų intervalą, veiks gan ilgai. Vykdymo laikas labai pagreitės, jei tyrinėsime tik tokias skaičiavimo sistemas, kuriose galioja nelygybė

0 < s div (1 + p + p2) < p 

Nėra sunku surasti šiuos rėžius. Akivaizdu, kad jei p = s, tai

s div (1 + p + p2) = 0

Lentelėje pritaikę dvejetainę paiešką randame tokią skaičiavimo sistemą p, kad

s div (1 + p + p2) > 0

bet s div (1 + (p+1) + (p+1)2) = 0. Šias sąlygas galima perrašyti taip:

s < (1 + (p+1) + (p+1)2) ir  s ( (1 + p + p2)

Perrinkimą pradedame nuo sistemos su šiuo pagrindu ir mažindami p tikriname tol, kol sąlyga s div (1 + p + p2) ( p tampa teisinga. Sistemų su mažesniu pagrindu nėra prasmės nagrinėti, nes jose skaičius s užsirašys keturiais ar daugiau skaitmenų.



const MAX   = 20;    { maksimalus ieškomų skaičių kiekis }

      MAX_N = 10000; { maksimali n reikšmė }

type  rezult = array [1..MAX] of

                record

                  skaičius,   { skaičius dešimtainėje sistemoje }

                  sis1, sis2, { abi sistemos }

                  sk1, sk2: longint 

                  { skaitmenys atitinkamose sistemose }

                 end;

var lent: array [2..MAX_N] of longint;

        { lentelė, kurioje bus saugomas skaičiaus 111, užrašyto visose }

        { skaičiavimo sistemose nuo 2 iki MAX_N, atitikmuo dešimtainėje }

        { sistemoje }

        { lent – globalus kintamasis }



procedure vienodi (m, n: longint;

                   var rez: rezult;

                   var kiek: longint); 

                   { kiek rasta įdomių skaičių }

  var  rasta, { keliomis sistemomis trimis vienodais skaitmenimis }

              { užsirašo skaičius }

       pag1, pag2, { abiejų sistemų pagrindai }

       s           : longint;

  procedure lentelė;

  { sudaro lentelę, kurioje bus saugomas skaičiaus 111, užrašyto visose }

  { skaičiavimo sistemose nuo 2 iki MAX_N, atitikmuo dešimtainėje }

  { sistemoje }

  { kintamasis lent – globalus }

    var i : longint;

  begin

    for i := 2 to MAX_N do

      lent[i] := 1 + i + i * i

  end; { lentelė }



  function ieškoti (x: longint): longint;

  { dvejetainės paieškos algoritmas }

  { rezultatas – lentelės lent indeksas i, }

  { toks kad lent[i - 1] <= x < lent[i] }

    var i1, i2, i: longint;

  begin

    i1 := 2;

    i2 := x;

    while i1 < i2 do

      begin

        i := (i1 + i2) div 2;

        if x < lent[i]

           then i2 := i

           else i1 := i + 1

       end;

    ieškoti := i1

  end; { ieškoti }



begin { vienodi }

  lentelė;     { sudarome lentelę }

  kiek := 0;   { dar nerasta nei vieno skaičiaus }

  if m = 1     { vieneto tikrai negalėsime užrašyti trimis skaitmenimis }

     then m := 2;

  for s := m to n do  { nagrinėsime visus skaičius iš duoto intervalo }

    begin

      rasta := 0;

      pag1 := ieškoti (s); { ieškoma didžiausio galimo pagrindo }

      { ieškosime pirmos sistemos, kurioje skaičius užsirašo }

      { trimis vienodais skaitmenimis }

      while (rasta = 0) and (pag1 > 2) do

        begin

          pag1 := pag1 - 1;

          if (s div lent[pag1] >= pag1) 

            { t.y. skaičiavimo sistemose }

            { su mažesniais pagrindais skaičius s užsirašys keturiais }

            { ar daugiau skaitmenų; tų atvejų nereikia tyrinėti }

            then pag1 := 1

            else if (s mod lent[pag1] = 0)  

                    { skaičius užsirašo trimis vienodais skaitmenimis }

                    then rasta := rasta + 1 

        end;

      { ieškosime antrosios sistemos }

      pag2 := pag1;

      while (rasta = 1) and (pag2 > 2) do

        begin

          pag2 := pag2 - 1;

          if (s div lent[pag2] >= pag2)

             then pag2 := 1

             else if (s mod lent[pag2] = 0)

                     then rasta := rasta + 1

        end;

      if rasta = 2  { jei radome dvi sistemas }

         then begin { įsimename rezultatą }

                kiek := kiek + 1;

                rez[kiek].skaičius := s;

                rez[kiek].sis1 := pag1;

                rez[kiek].sk1 := s div lent[pag1];

                rez[kiek].sis2 := pag2;

                rez[kiek].sk2 := s div lent[pag2];

              end;

    end;

end;



Intervale nuo 1 iki 10000 yra aštuoni tokie skaičiai

273

16 1 1 1

9 3 3 3�27310 = 1 1 116 = 3 3 39;��546

16 2 2 2

9 6 6 6�54610 = 2 2 216 = 6 6 69;��931

30 1 1 1

11 7 7 7�93110 = 1 1 130 = 7 7 711;

��3549

22 7 7 7

16 13 13 13

�54610 = 2 2 216 = 6 6 69;

��3783

61 1 1 1

35 3 3 3�378310 = 1 1 161 = 3 3 335;

��4557

67 1 1 1

25 7 7 7�455710 = 1 1 167 = 7 7 725;

��7566

61 2 2 2

35 6 6 6�755610 = 2 2 261 = 6 6 635;

��9114

67 2 2 2

25 14 14 14�911410 = 2 2 267 = 14 14 1425;

��

Žemiau pateikiame lentelę su testais:

Testo nr.�Pradiniai duomenys��1�544 547��2�300 310��3�1 1000��4�1 10000��

II dalis

Jaunesniems (iki IX kl.)

131. Lobio ieškojimas. Šį uždavinį galima būtų pavadinti labiau skaitymo, nei algoritmavimo uždaviniu. Sąlyga labai ilga ir paini. Tačiau ją įsiskaitęs pamatai, kad sprendimas nėra kažkuo ypatingas. Duotas labirintas, reikia jame rasti lobį ir iš jo išeiti. Įprasta labirintų uždavinius spręsti Bangos arba Paieškos į plotį metodu. Deja, šiam uždaviniui tų metodų pritaikyti negalime. „Reikalą gadina“ dvi sienos ir sprogmenys. Galima rasti trumpiausią kelią nuo įėjimo iki lobio ir nuo lobio iki išėjimo. Tačiau, ką daryti, jei ateinant trumpiausiu keliu teko išsprogdinti dvi skyles sienose, o išeinant trumpiausiu keliu taip pat tenka sprogdinti sienas kitose vietose. 

Todėl pasirenkame Grįžimo metodą. Labirintas yra nedidelis, taigi laiko limitas nebus viršyti. 

Apeinant labirintą grįžimo metodu, žymimasi kuriuose langeliuose būta. Mes taip pat žymėsime kuriuose langeliuose būta. Tačiau radus lobį, labirinto langelių būsenas teks įsiminti, ir vėl pažymėti, kad nei viename labirinto langelyje dar nebūta: juk grįžtant gali tekti eiti tais pačiais langeliais kaip ir atėjome. Įsiminti aplankytus langelius lobio atradimo momentu reikia tam, kad galėtume grįžti atgal.

Įsimenant kurie langeliai buvo aplankyti, svarbu įsiminti, kaip į juos pakliuvome: ar atėjome patys, ar mus permetė (atnešė). Imkime langelį D2 (131.2 pav.).

131.2 pav.

Jei į šį langelį ateiname patys, tai iš jo mus permeta į D3. Jei į šį langelį, pakliūname iš D1, galime pasiekti lobį. 

Sprendimą pradėkime nuo sienų formavimo. Mums būtų patogu, jei turėdami dviejų langelių koordinates iš karto galėtume pasakyti: skiria juos siena, ar ne. Tam sudarykime keturių matavimų loginę lentelę. Du pirmieji elemento indeksai reikš pirmojo langelio koordinates, du paskutinieji – antrojo. 

type sienos = array [1..D, 1..D, 1..D, 1..D] of boolean;

Toks vaizdavimas užima daug atminties. Kadangi labirintas labai mažas tai sprendimui nekliudys.

type siena = record

               kuri, { su kuria išorine siena liečiasi }

               n1, n2: integer;  { kuriuos stulpelius (eilutes) skiria }

             end;

     sienos = array [1..D, 1..D, 1..D, 1..D] of boolean;



procedure form_sienas (s2, s3: siena; var s: sienos);

{ suformuoja sienas }

  var y, x, u, v, i, kryptis: integer;

begin

  for y := 1 to D do

    for x := 1 to D do

      for u := 1 to D do

        for v := 1 to D do

            s[y, x, u, v] := false;

  for i := 1 to 5 do

    begin

      if i <= 2  { prie kurios labirinto sienos glaudžiasi ši siena }

         then kryptis := s2.kuri

         else kryptis := s3.kuri;

      if i in [1, 3] { randame stulpelį (eilutę) }

         then if (kryptis = 1) or (kryptis = 4)

                 then y := 1

                 else y := 4

         else if (kryptis in [1, 4]) and (i in [2, 4]) or

                 (kryptis in [2, 3]) and (i = 5)

                 then y := 2

                 else y := 3;

      if (kryptis = 1) or (kryptis = 3)

         then if i <= 2

                 then s[y, s2.n1, y, s2.n2] := true

                 else s[y, s3.n1, y, s3.n2] := true

         else if i <= 2

                 then s[s2.n1, y, s2.n2, y] := true

                 else s[s3.n1, y, s3.n2, y] := true;

    end;

  for y := 1 to D do

   for x := 1 to D do

    for u := 1 to D do

      for v := 1 to D do

        if s[y, x, u, v]

           then s[u, v, y, x] := true

end; { form_sienas }



Žemiau pateikiame patį algoritmą:

const D = 4; { labirinto dydis }

type  labirintas = array [1..D, 1..D] of

         record

           t: char;     { laukelio tipas }

           nr: integer; { laukelio numeris }

           būta_s,          { ar pakliūta į langelį savarankiškai }

           būta_n: boolean; { ar pakliūta į langelį savarankiškai }

         end;



procedure nebūta (y, x: integer; var l: labirintas);

{ pažymi, kad nebūta nei viename labirinto laukelyje, }

{ išskyrus laukelį su koordinatėmis (y, x) }

  var i, j: integer;

begin

 for i := 1 to D do

   for j := 1 to D do

     begin

       l[i, j].būta_s := false;

       l[i, j].būta_n := false;

     end;

 l[y, x].būta_s := true;

end; { nebūta }



function ėjimai (l: labirintas; { labirintas }

                 įėjim_y, įėjim_x: integer; { įėjimas }

                 s2, s3: siena { sienos }): integer;

{ randa lobį ir išeina iš labirinto }

  var dx, dy: array [1..4] of -1..1; { iš laukelio einama keturiomis }

                                     { kryptimis }

      min: integer;  { minimalus ėjimų skaičius }

      s: sienos;

  { visi šie kintamieji globalūs rekursinei procedūrai eiti }



  procedure rasti (y, x: integer; var kur_y, kur_x: integer);

  { randa langelį į kurį mus permeta (nuneša) iš langelio (y, x) }

    var tipas: char;

        ska, u, v: integer;

  begin

    { rasime į kokio tipo langelį mus perkels }

    if l[y, x].t = 'D'

       then tipas := 'D'

       else if (l[y, x].nr <= 2)

               then tipas := 'U'

               else tipas := 'Ž';

    { rasime koks bus to langelio numeris }

    if tipas = 'D'

       then if l[y, x].nr <= 2

               then ska := l[y, x].nr + 1

               else ska := 1

       else if tipas = 'Ž'

               then ska := 1

               else ska := l[y, x].nr + 3;

    { rasime tą langelį }

    for u := 1 to D do

      for v := 1 to D do

        if (l[u, v].t = tipas) and (l[u, v].nr = ska)

           then begin

                  kur_y := u; kur_x := v;

                end;

  end; { rasti }



  procedure eiti (y, x,   { langelio koordinatės }

                  sprogm, { turimų sprogmenų kiekis }

                  ėj: integer; { padarytų ėjimų skaičius }

                  rasta,  { ar rastas lobis }

                  pats: boolean; { ar pats atėjo į langelį }

                  var min: integer { minimalus ėjimų skaičius });

  { rekursinė procedūra, vaikščiojanti po labirintą }

    var k, u, v, kur_y, kur_x: integer;

        lab: labirintas;

        sprogo: boolean;

  begin

    if (y = įėjim_y) and (x = įėjim_x) and rasta

    then begin  { radome lobį ir pasiekėme išėjimą }

           if min > ėj

              then min := ėj;

           exit;

         end;

   if (l[y, x].t in ['U', 'D']) and pats

   then begin { jei iš šio laukelio mus nuneš(permes) į kitą }

          rasti (y, x, kur_y, kur_x);

          if not l[kur_y, kur_x].būta_n { jei jame nebūta }

          then begin

                 l[kur_y, kur_x].būta_n := true;

                 eiti (kur_y, kur_x, sprogm, ėj, rasta,

                       false, min);

                 l[kur_y, kur_x].būta_n := false

               end;

        end

   else begin { eisime visomis keturiomis kryptimis }

         if l[y, x].t = 'L' { jei pasiekėme lobį }

            then begin

                   lab := l;      { įsimename aplankytus laukelius }

                   rasta := true; { tikrai radome lobį }

                   nebūta (y, x, l); { vėl galime visur vaikščioti }

                 end;

         for k := 1 to 4 do

           begin

             u := y + dy[k]; { paeiname atitinkama kryptimi }

             v := x + dx[k];

             if (u in [1..D]) and (v in [1..D]) and 

                (ėj+1 < min)

                { jei langelis priklauso labirintui ir verta eiti }

             then if not l[u, v].būta_s { jei nebuvome šiame langelyje }

                  then begin

                         { gal reikia sprogdinti sieną }

                         sprogo := s[y, x, u, v];

                         if sprogo { jei reikia sprogdinti }

                         then begin

                                sprogm := sprogm - 1;

                                s[y, x, u, v] := false;

                                ėj := ėj + 1;

                              end;

                         if sprogm >= 0 { jei užteko sprogmenų }

                         then begin

                                l[u, v].būta_s := true;

                                eiti (u, v, sprogm, ėj+1, 

                                      rasta, true, min);

                                l[u, v].būta_s := false;

                              end;

                        if sprogo { grįžtame per žingsnį atgal }

                        then begin

                               sprogm := sprogm + 1;

                               s[y, x, u, v] := true;

                               ėj := ėj - 1;

                             end;

                       end;

           end;

         if l[y, x].t = 'L' { jei buvome radę lobį }

         then begin

                l := lab;  { atstatome buvusią labirinto padėtį }

                rasta := false; { lobio vėl neturime }

              end;

     end;

  end; { eiti }



    var  i, j: integer;

  begin { ėjimas }

    { apibrėžiame ėjimo kryptis }

    dy[1] := 0;  dx[1] := 1;

    dy[2] := 0;  dx[2] := -1;

    dy[3] := 1;  dx[3] := 0;

    dy[4] := -1; dx[4] := 0;

    { nė viename laukelyje dar nebūta }

    nebūta (įėjim_y, įėjim_x, l);

    min := maxint; { minimalus ėjimų skaičius }

    { suformuojame sienas }

    form_sienas (s2, s3, s);

    eiti (įėjim_y, įėjim_x, 2, 1, false, true, min);

    ėjimai := min + 1;   { išėjimas iš labirinto taip pat yra ėjimas }

  end; { ėjimai }



Testo nr.�Rezultatas�Paaiškinimai��1�7�Galima nesprogdinti nei vienos sienos (131.2 pav.)��2�9�Išnaudojami abu sprogmenys (131.3 pav.)��3�9�Išnaudojamas vienas sprogmuo (131.4 pav.)��4�6�Galima nenaudoti nei vieno sprogmens (131.5 pav.)��5�10�Išnaudojami abu sprogmenys (131.6 pav.)��6�8�Sutampa su sąlygoje pateiktu pavyzdžiu (131.1) pav.��

Testai iliustruojami žemiau pateiktuose paveikslėliuose.

131.3-6 pav.



132. Netvarkingas bibliotekininkas. Iš pažiūros algoritmas nėra sunkus: gan aišku, kaip atrasti vienokią ar kitokią klaidą ir kaip ją ištaisyti. Tačiau gali būti įvairiausių klaidų atvejų, jos persipina vienos su kitomis, algoritmas tampa ilgas, painus ir ... kai kuriais atvejais klaidingas. Autoriai nemažai vargę prie šio uždavinio nėra tikri, kad numatė absoliučiai visus atvejus... 

Algoritmą suskaidysime į dalis ir kiekvieną dalį aiškinsime atskirai. Uždavinio sprendimas skyla į dvi dalis: pirma pagal duotas lenteles bandoma dėlioti knygas į lentynas įsimenant ką ir kaip pavyko išdėlioti, po to remiantis turimais duomenimis taisomos klaidos.

Pradėkime nuo tipų aprašų bei globalių kintamųjų:

const MAX_L = 100;  { maksimalus leidinių skaičius }

      MAX_P = 2000; { maksimalus pozicijų skaičius }

type lentelė = array [1..MAX_L] of integer;

     lentyna = array [1..MAX_P] of integer;

     duomenys = array [1..MAX_L]

              of record

                   kiek, { kiek leidinio tomų pavyko išdėlioti }

                   pask: integer; { paskutinio padėto tomo vieta }

                 end;

{ žemiau surašyti globalieji kintamieji }

var n,  { leidinių skaičius }

    pozicijos: integer; { pozicijų skaičius lentynoje }

    vieta, skaič: lentelė; { pirmoji lentelė }

    perr2: lentyna;        { antroji lentelė }



Visi šio algoritmo pradiniai duomenys pateikti globaliais kintamaisiais. Jų reikšmės darbo metu nekeičiamos, be to jų yra nemažai ir jų reikia vos ne kiekvienai nedidelei procedūrai ar funkcijai, todėl pamanyta, kad bus patogiau jei nereikės kiekvienai procedūrai perdavinėti krūvos duomenų.

Atstatyti yra svarbiausia procedūra, sprendžianti uždavinį. Detalizuosime procedūrą ir joje esančius kreipinius į kitas procedūras.

Perrašinėjant lentelę gali būti padaryta įvairiausių klaidų, tačiau viena jų ypač nemaloni: tai nuoroda į neegzistuojančią poziciją. Pavyzdžiui, lentyna turi 15 vietų knygoms, o klaidingai perrašytame langelyje įrašytas skaičius 20. Klaida nemaloni dėl to, kad daugelyje situacijų tektų tikrinti ar neišeinama už lentelės ribų. Todėl pirmasis veiksmas – patikrinti, ar nėra nuorodos už lentelės ribų ir jei taip, šią klaidą pakeisti kita: vietoj netinkamo skaičiaus įrašyti nulį. Šios klaidos ieško funkcija už_ribų.

Atsižvelgus į šią klaidą dėliojame knygas. Dėliojimo rezultatas – dvi lentelės. Knygas dėliojame lentelėje lent. Imame po vieną leidinį ir dedame to leidinio tomus, kol jie pasibaigia (pasiekiame -1). Nustojame dėti leidinio tomus ir tada, kai randame klaidą (reikia dėti knygą į langelį, kuriame įrašytas 0 arba dėti knygą į vietą, kurioje jau padėta knyga). 

��                 l, jei toje vietoje padėtas l-tojo leidinio tomas;

lent[i] (    0, jei šioje vietoje antrojoje lentelėje buvo įrašytas 0;

                 -2, jei šioje vietoje nepadėta knyga;

Beje, dėdami knygas, netikriname, ar padėtų tomų skaičius lygus leidinio tomų skaičiui. 

Sąlygoje pateiktam antram pavyzdžiui ši lentelė atrodytų taip:

Vieta lentynoje�1�2�3�4�5�6�7�8�9�10�11�12�13�14��Leidinys�1�1�2�1�-2�2�2�3�2�3�4�0�0�0��

Lentelėje leid įsimenami duomenys apie kiekvieną leidinį: kiek jo tomų pavyko išdėlioti lentynoje ir kokioje vietoje yra paskutinis sėkmingai padėtas tomas.

Išdėliojus knygas, einama prie klaidų paieškos. Klaidų analizę pateiksime vėliau. Tada ir paaiškinsime, kodėl pakanka patikrinti dvi sąlygas norint sužinoti, ar lentelė perrašyta teisingai. Nustačius, kad yra klaida, kreipiamasi į procedūras ieškoti_skaičiaus ir ieškoti_vietos, kurios ieško klaidos pozicijos ir teisingos reikšmės toje pozicijoje.

procedure atstatyti (var yra: boolean; 

                     var kl_vieta, kl_sk: integer);

{ patikrina, ar antroje lentelėje yra klaida ir jei galima, ją atstato }

  var l, poz: integer;

      leid: duomenys;

      lent: lentyna;

begin

  { jei yra nuorodą į neegzistuojančią poziciją, šią klaidą keičiame kita }

  poz := už_ribų;

  if poz > 0

     then perr2[poz] := 0;

  { išdėliojame knygas }

  dėlioti (leid, lent);

  yra := false;

  for l := 1 to n do

    if (leid[l].kiek <> skaič[l]) or

       (perr2[leid[l].pask] <> -1)

       then yra := true

       else begin

              ieškoti_skaičiaus (kl_sk);

              ieškoti_vietos (lent, leid, kl_vieta)

            end;

end; { atstatyti }



Funkcija už_ribų bei procedūra dėlioti yra akivaizdžios: jų nekomentuosime.

function už_ribų: integer;

{ tikrina, ar nėra nuorodos į neegzistuojančią poziciją }

  var i, poz: integer;

begin

  poz := 0;

  for i := 1 to pozicijos do

    if (perr2[i] < -1) or (perr2[i] > pozicijos)

        then poz := i;

  už_ribų := poz

end; { už_ribų }





procedure dėlioti (var leid: duomenys; 

                   var lent: lentyna);

{ bando dėlioti knygas lentynoje }

{ leid – kiekvieno leidinio išdėliotų tomų skaičius ir paskutinio tomo vieta, }

{ lent – rodo kurioje pozicijoje kurio leidinio tomai yra }

  var tomas, buvo, l, i, poz: integer;

begin

  { inicializuojame masyvus leid ir lent }

  for l := 1 to n do

    begin

      leid[l].kiek := 0;

      leid[l].pask := 0;

    end;

  for i := 1 to pozicijos do

    if perr2[i] = 0

       then lent[i] := 0

       else lent[i] := -2;   { nei viena pozicija lentynoje neužimta }

  { dėliosime kiekvieno leidinio tomus }

  for l := 1 to n do

    begin

      buvo := 0;           { buvusi pozicija }

      poz := vieta[l];     { pirmojo tomo vieta }

      tomas := 0;          { kiek tomų sudėliota }

      while (poz <> -1) and (poz <> 0) do

        { kol nepriėjome paskutinio tomo ar nuorodos į nulinę poziciją }

        if (poz > 0)

        then if lent[poz] <> -2 { priėjome užimtą vietą }

             then poz := 0

             else begin

                   lent[poz] := l; { šioje vietoje yra leidinys l }

                   tomas := tomas + 1;

                   buvo := poz;

                   poz := perr2[poz];

                  end;

      { įsimename duomenis apie l-tąjį leidinį }

      leid[l].kiek := tomas;

      leid[l].pask := buvo;

    end

end; { dėlioti }



Čia baigiasi pirmoji algoritmo dalis: turėdami reikiamus duomenis pereiname prie klaidų analizės. 

Procedūra ieškoti_skaičiaus randa koks skaičius turėjo būti įrašytas klaidingai perrašytame langelyje.

Perrašant lentelę buvo padaryta klaida: vienas skaičius pakeistas kitu skaičiumi. Tarkime, kad klaidingai perrašytoje pozicijoje buvo įrašytas skaičius sk1. Perrašant suklysta ir jis pakeistas kitu skaičiumi. 

Iš sąlygos žinome, kad nulis negalėjo būti pakeistas kitu skaičiumi, t. y. sk1 ( 0. Jei sk1 ( -1, tai perrašytoje lentelėje minus vienetų skaičius nebelygus leidinių skaičiui. 

Liko atvejis, kai sk1 > 0. Visi natūralieji skaičiai antroje lentelėje reiškia lentynos poziciją. Į kiekvieną lentynos poziciją (išskyrus tas kurios yra tuščios) turi būti nuoroda, t. y. tos pozicijos numeris turi būti įrašytas arba pirmoje arba antroje lentelėje. Jei sk1 pakeičiamas kitu skaičiumi, tuomet į poziciją sk1 nebėra nuorodos, t. y. šios pozicijos numeris nėra įrašytas nei pirmoje, nei perrašytoje lentelėje.

Remdamiesi šiais samprotavimais rašome procedūrą ieškoti_skaičiaus. Joje yra kreipiniai į funkciją kiek ir procedūrą nuoroda. Kiek(x) suskaičiuoja kiek kartų skaičius x pasikartoja perrašytoje lentelėje. Procedūra nuoroda skaičiuoja kiek kartų pirmoje ir antroje (perrašytoje) lentelėse sutinkama nurodyta pozicija.

procedure ieškoti_skaičiaus (var kl_sk: integer);

{ ieško teisingos reikšmės }

  var i, sk, n1, n2: integer;

begin

  if kiek(-1) < n

     then kl_sk := -1

     else begin

            for i := 1 to pozicijos do

              begin

                if perr2[i] <> 0 

                   then nuoroda (i, sk, n1, n2);

                if sk = 0 then kl_sk := i

              end;

          end;

end; { ieškoti_skaičiaus }



Pateikiame funkciją kiek.

function kiek (x: integer): integer;

{ kiek kartų skaičius x pasikartoja perrašytoje lentelėje }

  var i, sk: integer;

begin

  sk := 0;

  for i := 1 to pozicijos do

    if perr2[i] = x

       then sk := sk + 1;

  kiek := sk

end; { kiek }



Keletas komentarų apie procedūrą nuoroda. Ji randa ne tik nuorodų skaičių į duotą poziciją x, bet ir pačias nuorodas, t. y. antrosios lentelės langelių numerius, kuriuose įrašytas x. Padarius vieną klaidą antroje lentelėje tokių nuorodų gali būti ne daugiau kaip dvi. 

Beje, jei į poziciją x rodoma iš pirmosios lentelės, tuomet ta nuoroda žymima neigiamu skaičiumi ir reiškia leidinio numerį, kurio pirmas tomas turi būti pozicijoje x. Kadangi pirmoji lentelė yra teisinga, iš jos gali būti tik viena nuoroda į x.

procedure nuoroda (x: integer;

                   var sk, { langelių skaičius }

                   n1,n2: integer { iš kurių langelių rodoma į x });

{ ieško kuriame langelyje įrašytas skaičius x }

{ nuorodą iš pirmosios lentelės žymėsime neigiamu skaičiumi }

  var poz, i, l: integer;

begin

  n1 := 0; n2 := 0;

  { tikrina, ar nėra nuorodos iš pirmosios lentelės }

  for l := 1 to n do

    if vieta[l] = x

       then n1 := -l; { nuorodą iš pirmosios lentelės }

               { susitarkime žymėti neigiamu skaičiumi }

  { tikrina, ar nėra tokios nuorodos antroje lentelėje }

  for i := 1 to pozicijos do

    if (perr2[i] = x)

       then if n1 = 0 then n1 := i

                      else n2 := i;

  sk := ord (n1 <> 0) + ord (n2 <> 0)

end; { nuoroda }



Radę teisingą reikšmę, bandysime ieškoti klaidos vietos. Pradžioje išanalizuosime galimas klaidas bei jų pasekmes.

Imkime sąlygoje pateikto pavyzdžio pirmuosius du leidinius. Pagal antrąją lentelę galima sudaryti tokias grandines:

1 leidinys:        1 –> 4  –> 2  –> 5  –> -1 

2 leidinys:        3 –> 6  –> 7  –> 9  –> -1 

Sudarysime galimų klaidų bei jų pasekmių lentelę. Daugtaškiu žymėsime nutrūkusią grandinę.

Klaidos apibūdinimas�Nr�Pavyzdžiai�Pasekmė��Skaičius pakeičiamas 0�1�1 –> 4 –> 0  ....–> 5  –> -1 �Ketvirtame langelyje įrašius 0, nulių skaičius tampa vienetu didesnis, nei turi būti, atsiranda nuoroda į nulį (pirmasis langelis nukreipia į ketvirtą, kuriame įrašytas 0), t. y. grandinė baigiasi nekorektiškai (ne -1, pirmojo leidinio sėkmingai išdėliojami ne visi tomai, lieka nutrūkęs grandinės galas;���2�0 –> 4 –> 2 .–> 5  –> -1 �Tas pats, kas praeitame pvz, tik nepadedamas nei vienas tomas;.���3�1 –> 4 –> 2 .–> 5  –> 0 �Tas pats, kas praeitame pvz, tik sumažėja -1 skaičius ir nėra grandinės galo��Skaičius pakeičiamas -1�4�1 –> 4 –> -1... .–> 5  –> -1 �Padidėja -1 skaičius, lieka nutrūkęs grandinės galas, išdėliojami ne visi leidinio tomai;���5�-1... –> 4 –> 2 .–> 5  –> -1 �Tas pats, tik nepadedamas nei vienas tomas;��Skaičius keičiamas skaičiumi�����grandinė užsiciklina�6�1 –> 4 –> 2 .–> 4 .... –> -1 �Nepavyksta išdėlioti visų tomų, lieka nutrūkęs grandinės galas, grandinė baigiasi nekorektiškai (ne -1), į kažkurią poziciją egzistuoja dvi nuorodos;���7�1 –> 4 –> 2 .–> 5  –> 4 �Visi tomai išdėliojami, tačiau padidėja -1 skaičius ir grandinė baigiasi nekorektiškai, į kažkurią poziciją egzistuoja dvi nuorodos;��nuoroda į nulinį langelį�8�1 –> 14... –> 2 .–> 5  –> -1 �Lieka nutrūkusi grandinė, išdėliojami ne visi tomai, grandinė baigiasi nekorektiškai, atsiranda nuoroda į nulį, ���9�1 –> 4 –> 2 .–> 5  –> 14 �Nelieka nutrūkusios grandinės, išdėliojami visi tomai, tačiau grandinė baigiasi nekorektiškai, atsiranda nuoroda į nulį��Susilieja dvi grandinės�10�1 –> 4 –> 2 .–> 5  –> -1 

 3 –> 2....  –> 7  –> 9  –> -1�Jei pirma dedami pirmojo leidinio tomai:

Lieka nutrūkusi grandinė, pirma grandinė išdėliojama korektiškai, antra – nekorektiškai, į kažkurią poziciją egzistuoja dvi nuorodos;; ���11��Jei pirma dedami antrojo leidinio tomai:

Lieka nutrūkusi grandinė, pirma grandinė išdėliojama nekorektiškai, antra korektiškai, tačiau išdėliojama per mažai antrosios tomų, į kažkurią poziciją egzistuoja dvi nuorodos;���12�1 –> 4 –> 2 .–> 5  –> 3 

 3.. –> 6  –> 7  –> 9  –> -1 �Lieka nutrūkusi grandinė, pirma grandinė išdėliojama nekorektiškai, antrosios – nepadedamas nei vienas tomas, į kažkurią poziciją egzistuoja dvi nuorodos;���13�1 –> 4 –> 7 .–> 5  –> -1 

3 –> 6  –> 7 ... –> 9  –> -1 

�Lieka nutrūkusi grandinė, viena grandinių išdėliojama nekorektiškai. Tačiau bendros abiejų grandinių dalies ilgis lygus pamestos grandinės ilgiui, todėl nustatyti klaidos vietos neįmanoma, į kažkurią poziciją egzistuoja dvi nuorodos;��Pametamas grandinės vidinė dalis�14�1 –> 4  –> 5  –> -1

       2  –> 5  �Lieka pamesta grandinės dalis, padėtų tomų skaičius nelygus leidinio tomų skaičiui, egzistuoja dvi nuorodos į tą patį laukelį��

Kiekviena klaida sukelia įvairiausių pasekmių (lieka nutrūkęs grandinės galas, nekorektiškai baigiasi grandinė, padidėja nulių skaičius ir t. t.). Klaidas galima įvairiausiai klasifikuoti pagal pasekmes, tas pasekmes identifikuoti ir pagal tai ieškoti klaidos vietos. Mes pasirinkome vieną variantą, kuris mums atrodė priimtiniausias. 

Pirmiausia atkreipkite dėmesį, kad kokia bebūtų padaryta klaida, arba nors viena grandinė baigsis nekorektiškai arba kurio nors leidinio išdėliotų tomų skaičius nebus lygus tikrajam to leidinio tomų skaičiui. Užtenka patikrinti šias dvi sąlygas, kad pasakytum, ar perrašant lentelę buvo padaryta klaida, ar ne. Tai ir daroma procedūroje atstatyti. 

Procedūroje ieškoti_vietos klaidas suskirstėme į tris grupes:

1) yra nuoroda į nulinį elementą (1, 2, 3, 8, 9 pavyzdžiai);

2) į kažkurią poziciją egzistuoja dvi nuorodos (6, 7, 10, 11, 12, 13, 14) pavyzdžiai;

3) kai skaičius pakeičiamas minus vienetu (4, 5 pavyzdžiai).

Šios trys grupės nepersikerta ir išsemia visus galimus atvejus.

procedure ieškoti_vietos (lent: lentyna; leid: duomenys;

                          var kl_vieta: integer);

{ ieško pozicijos, kurioje padaryta klaida }

  var poz, i, l, sk, n1, n2: integer;

begin

  kl_vieta := 0;

  { gal būt yra nuoroda į nulinį elementą }

  ieškoti0 (kl_vieta);

  if kl_vieta = 0   { jei tokios klaidos nerasta }

     then begin     { ieškome kitų klaidų }

            { gal ta pati pozicija įrašyta į du langelius }

            sk := 0; i := 1;

            while (sk <> 2) and (i <= pozicijos) do

              begin

                if perr2[i] <> 0 

                   then nuoroda (i, sk, n1, n2);

                i := i + 1;

              end;

            if sk = 2

             then du_langeliai(lent,leid,n1,n2,kl_vieta)

             else galas (leid, kl_vieta);

          end;

end; { ieškoti_vietos }



Imame pirmąjį atvejį: nuoroda į nulinį elementą. Atkreipkite dėmesį, kad šiuo atveju terminas nuoroda suprantamas kitaip. Tai reiškia ne langelio, kuriame įrašytas nulis, numerį, bet langelio, kuriame įrašytas pozicijos (kurioje įrašytas nulis) numeris. Pirmiausia reikia patikrinti, ar yra nuoroda į nulį. Klaida gali būti padaryta dvejopa. Imkime sąlygoje pateiktą antrąją lentelę:

1�2�3�4�5�6�7�8�9�10�11�12�13�14��4�5�6�2�-1�7�9�-1�-1�8�-1�0�0�0��

Joje galime skaičių pakeisti nuliu (klaidingai perrašytas langelis patamsintas):

1�2�3�4�5�6�7�8�9�10�11�12�13�14��4�5�6�0�-1�7�9�-1�-1�8�-1�0�0�0��

arba galime vietoj nuorodos (ar minus vieneto) įrašyti nuorodą į nulinį langelį:

1�2�3�4�5�6�7�8�9�10�11�12�13�14��4�5�14�2�-1�7�9�-1�-1�8�-1�0�0�0��

Pirmuoju atveju klaida bus tame langelyje, kuriame įrašytas nulis, antruoju – langelyje, kuriame yra nuoroda į nulį. Pirmuoju atveju padidėja nulių skaičius lentelėje, antruoju – ne.

Procedūra ieškoti0 pirmiausia turi surasti langelį, iš kurio yra nuoroda į nulį. Vienu atveju tai bus pirmasis langelis, kitu – trečiasis. Radus tokį langelį, pagal nulių skaičių lentelėje reikia nustatyti, ar tai pirmasis ar antrasis atvejis. 

procedure ieškoti0 (var kl_vieta: integer);

{ taiso klaidą - nuorodą į nulinį langelį }

{ kl_vieta = 0, jei tokia klaida nepadaryta }

  var poz, i, l, sk, turi0, x1, x2: integer;

begin

  poz := 0; { ieškome nuorodos į nulį iš pirmosios }

  for l := 1 to n do

    if perr2[vieta[l]] = 0

       then poz := -l;

  { ir iš antrosios lentelės }

  for i := 1 to pozicijos do

    if perr2[i] > 0     { jei tai nuoroda, o ne grandinės pabaiga }

       then if perr2[perr2[i]] = 0

               then poz := i;

  if poz <> 0   { jei nuoroda yra }

     then begin

            { rasime kiek nulių turi būti lentelėje }

            turi0 := pozicijos;

            for l := 1 to n do

               turi0 := turi0 - skaič[l];

            if turi0 = kiek(0)

               then { vietoj skaičiaus įrašytas skaičius, rodantis į nulį }

                 if poz < 0

                    then kl_vieta := vieta[-poz]

                    else kl_vieta := poz

               else if poz < 0  { vietoj skaičiaus įrašytas 0 }

                       then kl_vieta := vieta[-poz]

                       else kl_vieta := perr2[poz];

          end;

end; { ieškoti0 }



Einame prie antros klaidos: į tą pačią poziciją egzistuoja dvi nuorodos. Pažymėkime jas n1 ir n2. tai reiškia, kad klaidingai perrašytos lentelės pozicijose n1 ir n2 įrašytas tas pats skaičius. Akivaizdu, kad klaida padaryta, arba pozicijoje n1 arba pozicijoje n2.

Jei nors viena šių nuorodų yra iš pirmosios lentelės, matome, kad klaida padaryta antrojoje: juk pirmoji lentelė perrašyta teisingai. 

Jei ne, randame kurių leidinių tomai įrašyti langeliuose n1 ir n2. Pažymėkime juos l1 ir l2. Jei nors vienoje šių pozicijų nepadėtas joks leidinys, tai reiškia, kad buvo pamesta vienos grandinės dalis (14 pavyzdys) ir klaida padaryta toje pozicijoje, kurioje yra padėta knyga. 

Dar vis neradus klaidos, patikrinkime, ar sutampa l1 ir l2. Jei sutampa, tai reiškia, kad grandinė užsiciklino ir klaida padaryta toje pozicijoje, kur pavyko padėti šio leidinio tomą su didžiausiu numeriu (nebūtinai tai paskutinysis tomas).

Na ir liko paskutinysis atvejis, kai susilieja dvi grandinės. Jei pamestoji grandinės dalis lygi bendrajai abiejų grandinių daliai – klaidos vietos nustatyti neįmanoma. Jei tos dalys nelygios, žinodami kiek kiekvienas šių dviejų leidinių turi tomų, randame klaidos vietą.

procedure du_langeliai (lent: lentyna; leid: duomenys;

                        n1, n2: integer; 

                        var kl_vieta: integer);

{ taisoma klaida, kai į du langelius (n1 ir n2) įrašytas tas pats skaičius }

  var poz, i, pam, bendra, l1, l2: integer;

begin

  { pirmoje lentelėje klaidos negali būti }

  if n1 < 0  then kl_vieta := n2 { n1 – nuoroda iš pirmos lentelės }

             else if n2 < 0 then kl_vieta := n1;

  if kl_vieta <> 0 then exit;



  { rasime leidinius, kurių tomai šiose pozicijose }

  l1 := lent[n1]; l2 := lent[n2];

  { gal buvo pamestas vienos grandinės vidurys }

  if l1 = -2 then kl_vieta := n2

             else if l2 = -2 then kl_vieta := n1;

  if kl_vieta <> 0 then exit;



  { gal susidarė ciklas vienoje grandinėje }

  if l1 = l2

     then kl_vieta := leid[l1].pask;

  if kl_vieta <> 0 then exit;



  { jei ne, tada susiliejo dvi grandinės }

  { rasime pamestų pozicijų skaičių }

  pam := 0;

  for i := 1 to pozicijos do

    if lent[i] = -2

       then pam := pam + 1;

  { rasime bendrą abiejų grandinių dalį }

  bendra := 0;

  poz := perr2[n1];

  while poz <> -1 do

    begin

      bendra := bendra + 1;

      poz := perr2[poz];

    end;

  if bendra <> pam  { bendra dalis nelygi pamestai }

    then if (perr2[n1] = l1) and 

            (leid[l2].kiek+bendra = skaič[l2]) or

            (perr2[n2] = l2) and 

            (leid[l1].kiek+bendra <> skaič[l1])

             then kl_vieta := n2

             else kl_vieta := n1

    { else – atstatyti neįmanoma }

end; { du_langeliai }



Liko trečiasis atvejis: skaičius pakeistas minus vienetu. Tokiu atveju visų, išskyrus vieno, leidinių tvarkingai išdėliojami visi tomai. O tame leidinyje klaida padaryta pozicijoje, į kurią padėjus tomą užsibaigė grandinė.



procedure galas (leid: duomenys; var kl_vieta: integer);

{ taisoma klaida, kai pamestas vienos grandinės galas }

  var l: integer;

begin

  { rasime klaidingai perrašyto leidinio numerį }

  l := 1;

  while (l <= n) and (leid[l].kiek = skaič[l]) do

    l := l + 1;

  kl_vieta := leid[l].pask

end; { galas }



Testo nr.�Rezultatai�Paaiškinimai��1�GERAI�klaidos nepadaryta��2�11 -1�-1 pakeistas skaičiumi, rodančiu į nulinį langelį��3�BLOGAI�dvi grandinės susilieja; lieka vienodas skaičius neišdėliotų tomų, todėl atstatyti negalima��4�3 6�klaida vienoje grandinėje: pamesta jos vidurinė dalis��5�7 9�klaida vienoje grandinėje, ji ciklinasi (skaičius pakeistas kitu skaičiumi)��6�8 -1�klaida vienoje grandinėje, ji ciklinasi (-1 pakeistas kitu skaičiumi)��7�1 4�pirmo tomo pozicijoje esantis skaičius (nuoroda į antrąjį tomą) pakeistas minus vienetu ��8�4 2�skaičius pakeistas kitu skaičiumi, rodančiu į nulinį langelį��9�5 -1�susilieja dvi grandinės, tačiau klaidą atstatyti galima��10�3 6�dvi grandinės sulipdomos (t. .y vienos paskutinis elementas rodo į kitos pirmą ��11�8 -1�minus vienetas pakeistas nuliu��12�3 6�pirmo tomo pozicijoje esantis skaičius (nuoroda į antrąjį tomą) pakeistas nuliu ��13�4 2�nuoroda į neegzistuojančią lentynos poziciją��14�7 9�skaičius pakeistas nuliu��15�4 2�skaičius pakeistas minus vienetu��

133. Nežinomas kodavimas. Pradėkime nuo baito sąvokos. Baitą sudaro aštuoni bitai. Bitas – tai vienetas arba nulis.

Bito nr.�7�6�5�4�3�2�1�0���0�1�1�0�1�0�0�0��

Pavyzdžiui, baitas 011010002 ( 6816 ( 10410; Tai yra dvejetainis, šešioliktainis bei dešimtainis baito užrašai.

Į pradinių duomenų bylą žiūrima kaip į ištisinę bitų eilę. Fiziškai bylose bitai saugomi nuo mažiausio iki vyriausio. Aukščiau parodytas baitas byloje bus saugomas taip:

Bito nr.�0�1�2�3�4�5�6�7���0�0�0�1�0�1�1�0��

Taigi jei būtų grupuojama po 5 bitus, į pirmąją grupę pakliūtų šie bitai: 0 0 0 1 0.

Sąlygoje pasakyta, kad bitai skirstomi į vienodo ilgio grupes po 7, 6, 5, 4, 3, arba 2 bitus. Taigi pirmiausia reikėtų nustatyti, į kokio ilgio grupes skirstomi bitai. Tam reikėtų susikurti keletą skirtingo ilgio pradinių duomenų bylų. 

Pavyzdžiui, susikuriame bylą, kurios ilgis – keturi baitai. Joje bus 4 ( 8 ( 32 bitai. Pagal rezultatų bylos ilgį jau galima daryti išvadas:

Grupės dydis�Rezultatų .bylos ilgis ��2�16 pilnų grupių ( 1 baitas ( 16 baitų��3�(10 pilnų grupių + 1 nepilna grupė) ( 1 baitas ( 11 baitų��4�8 pilnos grupės ( 1 baitas ( 8 baitai��5�(6 pilnos grupės + 1 nepilna grupė) ( 1 baitas ( 7 baitai��6�(5 pilnos grupės + 1 nepilna grupė) ( 1 baitas ( 6 baitai��7�(4 pilnos grupės + 1 nepilna grupė) ( 1 baitas ( 5 baitai��

Pasižiūrėję į rezultatų bylą galime pasakyti, į kokio didumo grupes skirstomi bitai. Mūsų uždavinio atveju bitų grupės ilgis yra 4.

Jei kiekviena keturių bitų grupė koduojama vienu baitu, reikia išnagrinėti visas galimas keturių bitų grupes. Jų bus šešiolika: 0000, 0001, .... 1111. Reikia susikurti pradinių duomenų bylą, kurioje būtų visos įmanomos kombinacijos. 

Tokią bylą nesunkiai galima sukurti patiems. Yra įvairių būdų tai padaryti. Tam reikėtų panagrinėti simbolių kodus. Dirbdami Paskaliu, tai galite padaryti naudodami funkciją ord. Ši funkcija grąžina dešimtainį simbolio kodą. Šešioliktainius simbolių kodus galite sužinoti žiūrėdami bylų turinius su Nortono komanderiu arba DOS'o navigatoriumi.

Panagrinėję pamatysite, kad simbolių ‘0’..., ‘9’ šešioliktainiai kodai yra 3016, ..., 3916. (gausime visas bitų kombinacijas nuo 0000 iki 1001), o simbolių ‘J’, ‘K’, ‘L’, ‘M’, ‘N’, ‘O’ kodai yra 4A16, ...4F16 ( gausime visas likusias bitų kombinacijas).

Sudarę tokią duomenų bylą, ir ją užkodavę, galime pasakyti, kaip koduojama kiekviena bitų grupė:

Grupė

(jos dveje�tainis užrašas)�Simbolis iš kodų lentelės, atitinkantis baitą, kuriuo koduojama��0000�0��0001�1��0010�2��0011�3��0100�4��0101�5��0110�6��0111�7��1000�8��1001�9��1010�A��1011�B��1100�C��1101�D��1110�E��1111�F��

Pavyzdžiui, pradinėje duomenų byloje yra simbolis ‘G’. Jo šešioliktainis kodas yra 4716 ( 010001112. Pirmą baitų grupę sudarys baitai 0111 (koduojama 7), antrąją – 0100 (koduojama 4). Taigi simbolis ‘G’ bus užkoduotas ‘74’. 

Pastebėsime, kad kiekvienas simbolis koduojamas dviem simboliais – apsuktu to simbolio šešioliktainiu kodu (du skaitmenys). Taip gaunama todėl, kad fiziniame lygyje baitai išdėstyti nuo mažiausios iki vyriausios skilties.

Pateikiame programos kod.exe pradinį tekstą:



program kodavimas_4_bitais;



  const bitai = 4;         { bitų grupės ilgis }

        lent: string[16] = '0123456789ABCDEF';

              { kodavimui naudojami simboliai }

        max = 10000; { maksimalus rezultatų bylos dydis baitais }



  procedure klaida;

  { nutraukia programos darbą dėl įvedimo arba išvedimo klaidos }

  begin

    writeln ('Įvedimo/išvedimo klaida!');

    halt

  end; { klaida }



  procedure atidar_bylas (var fskait, fraš: file);

  { patikrina parametrus ir atidaro bylas }

  begin

    writeln;

    if paramcount <> 2 { jei nenurodyti tinkami parametrai }

       then begin

              writeln ('Naudojimas: ');

              writeln(' KOD.EXE  duom.dat rez.dat');

              writeln;

              halt

            end;

    assign(fskait, paramstr(1));

    {$I-}

    reset (fskait, 1);

    {$I+}

    if ioresult <> 0

       then klaida;

    assign(fraš, paramstr(2));

    {$I-}

    rewrite(fraš, 1);

    {$i+}

    if ioresult <> 0

       then klaida

  end; { atidar_bylas }



  procedure uždar_bylas (var fskait, fraš: file);

  begin

    close(fraš);

    close(fskait)

  end; { uždar_bylas }



  procedure kodavimas (var fskait, fraš: file);

  { užkoduoja duomenis }

    var duom, rez: array [1..max] of byte;

        dydis, { pradinių duomenų bylos dydis baitais }

        rsk,   { rezultatų bylos dydis baitais }

        perskait, i, poslinkis, kaukė: integer;

        bit4: byte; { keturių bitų grupė }

  begin

    { inicializuojame pradinių duomenų ir rezultatų masyvus }

    fillchar (duom, sizeof(duom), 0);

    fillchar (rez, sizeof(rez), 0);

    { randame pradinių duomenų kiekį baitais }

    dydis := filesize (fskait);

    { perskaitome pradinius duomenis }

    blockread (fskait, duom, dydis, perskait);

    if perskait <> dydis

       then begin

              uždar_bylas (fskait, fraš);

              klaida

            end;

    i := 1;

    poslinkis := 0;

    rsk := 0;

    kaukė := $0F; { 00001111 }

    while i <= dydis do

      begin

        rsk := rsk + 1;

        { paimam eilinius 4 bitus }

        bit4 := (duom[i] shr poslinkis) and kaukė;

        { juos užkoduojame vienu baitu kurį įrašome į rezultatų masyvą }

        rez[rsk] := ord (lent[bit4 + 1]);

        poslinkis := poslinkis + bitai;

        if poslinkis > 7 { paslenkama arba per 0 arba per 4 bitus }

           then begin

                  poslinkis := poslinkis - 8;

                  i := i + 1

                end

      end;

    { įrašome užkoduotus duomenis į rezultatų bylą }

    blockwrite (fraš, rez, rsk, dydis);

    if rsk <> dydis

       then klaida

  end; { kodavimas }



  var fskait, fraš: file;

begin

  atidar_bylas (fskait, fraš);

  kodavimas (fskait, fraš);

  uždar_bylas (fskait, fraš)

end.



Kadangi šis uždavinys buvo skirtas jaunesniems moksleiviams, pateiksime paprastesnį jo sprendimą. Atkoduodami į pradinių duomenų ir į rezultatų bylas žiūrime kaip į tekstines bylas. Iš pradinių duomenų bylos skaitysime pu du simbolius (tai bus du baitai) ir juos iškoduosime. 

Du perskaitytieji simboliai bus du šešioliktainiai skaitmenys, pavyzdžiui ‘F4’. Šiuos simbolius sukeisime vietomis (gausime ‘4F’), paversime skaitmenimis ir gautą šešioliktainį skaičių užrašysime dešimtaine sistema. 4F16 ( 7910. Tai yra užkoduoto simbolio kodas. Belieka į rezultatų bylą išvesti šį simbolį. 

procedure atkoduoti (var fduom, frez: text);

  var b1, b2, simb: char;

      kodas, sk1, sk2: integer;

begin

  while not eof (fduom) do

    begin

      { perskaitome du baitus, kurias buvo koduojamas vienas baitas }

      read (fduom, b1, b2);

      { šešioliktainius skaitmenis užrašysime dešimtainiais }

      if b1 in ['A'..'Z']

         then sk1 := ord(b1) - ord('A') + 10

         else sk1 := ord(b1) - ord('0');

      if b2 in ['A'..'Z']

         then sk2 := ord(b2) - ord('A') + 10

         else sk2 := ord(b2) - ord('0');

      kodas := sk2 * 16 + sk1;

      simb := chr (kodas);

      write (frez, simb)

    end

end; { atkoduoti }



Šio uždavinio sprendimui patikrinti užtektų vieno didelio testo, kuriame būtų surašyti visi galimi simboliai (t. y. simboliai kurių kodai priklauso intervalui [00, … FF]).



134. Šaudymas. Algoritmas gana paprastas – imame po vieną šūvį, tikriname, ar jis pateko į taikinį ir skaičiuojame balus. 

Sunkiausias dalykas – patikrinti, ar šūvis pateko į apskritimo vidų ar ant jo ribos. Iš geometrijos žinome apskritimo lygtį:

(x - a)2 + (y - b)2 ( r2;

čia (a, b) – apskritimo centro koordinatės, r – jo spindulys.

Mūsų atveju visų apskritimų centrai sutampa su koordinačių pradžia. Taigi apskritimo lygtis supaprastėja: x2 + y2 ( r2. Visi taškai (x, y) patenkantys į apskritimo vidų ar ant jo tenkina nelygybę: x2 + y2 ( r2.

Taigi turime šūvio koordinates (x, y). Patikriname, ar šūvis patenka į mažiausiąjį apskritimą, kurio spindulys r, jei ne – imame didesnįjį apskritimą, kurio spindulys 2 ( r ir tikriname, ar šūvis pateko į jį ir t. t. Jei patikrinome visus apskritimus ir šūvis nepateko nė į vieną iš jų, vadinasi, šauta už taikinio ribos.

program šaudymas;



  function pateko (x,y: real; { šūvio koordinatės }

                   r: integer { mažiausio apskritimo spindulys }): 

                                          integer;

                      

  { suskaičiuoja kiek balų gauta už šį šūvį }

    var j, balai: integer; { į kurį apskritimą pateko šūvis }

  begin

    j := 1;

    balai := 0;

    while j <= 10 do

      begin

        if sqrt (sqr(x) + sqr(y)) <= r

           then begin

                  balai := 11 - j;

                  j := 11

                end

           else  j := j + 1;

         r := r * 2 { kito apskritimo spindulys bus dukart didesnis }

      end;

    pateko := balai

  end; { pateko }



  var x, y,  { šūvio koordinatės }

      n, r,  { šūvių skaičius ir mažiausio apskritimo spindulys }

      balai, { surinkti balai }

      i: integer;

      f: text;

begin

  assign(f, 'ŠAUD.DAT');

  reset(f);

  readln(f, r, n);

  balai := 0;

  for i := 1 to n do

    begin

      readln (f, x, y);

      balai := balai + pateko (x, y, r)

    end;

  close(f);

  writeln(balai)

end.



Atkreipkite dėmesį į vieną svarbų dalyką. Pagrindinėje programos dalyje šūvių koordinatės – sveikieji skaičiai, procedūroje pateko – realieji. Taip parašėme dėl Turbo Paskalio ypatumų. Mat reiškinio sqrt(sqr(x) + sqr(y)) rezultatas yra realiojo tipo, tačiau keliant sveikuosius skaičius kvadratu ir juos sudedant gali įvykti perpildymas. Turbo Paskalis pirma atlieka veiksmus su sveikaisiais skaičiais ir gauna rezultatą – sveikąjį skaičių, o tik tada jį verčia realiuoju skaičiumi ir traukia šaknį.

Testo nr.�Šūvių skaičius teste�Rezultatas�Paaiškinimai��1�4�31�Vienas šūvis patenka ant apskritimo;��2�3�14�Vienas šūvis patenka ant apskritimo;��3�10�50�Tikrinant, ar šūvis pateko į apskritimą, gali įvykti perpildymas;��4�8�30�Tikrinant, ar šūvis pateko į apskritimą, gali įvykti perpildymas;��5�13�48�Vienas šūvis patenka ant apskritimo; tikrinant, ar šūvis pateko į apskritimą, gali įvykti perpildymas;��6�17�101�Vienas šūvis išeina už taikinio ribų;��7�20�129�Atsitiktinis testas��8�34�241�Vienas šūvis patenka ant apskritimo; vienas šūvis išeina už taikinio ribų;��9�101�796�Didesnis testas��10�25�231�Atsitiktinis testas��

135. Labirintas. Uždavinys – rasti išėjimą iš labirinto – yra klasikinis. Jam yra žinomi ir klasikiniai sprendimo metodai. Labirinto uždavinys dažniausiai sprendžiamas vienu iš šių būdų – grįžimo (arba paieškos į gylį) metodu, bangos algoritmu, paieškos į plotį metodu. 

Sprendimas, atrodo, gana akivaizdus. Pirmiausia reikia apeiti visą labirintą visomis kryptimis (t. y. įsiminti patį labirintą), o tuomet pritaikyti, pavyzdžiui, bangos metodą. Stop! Juk apeidami labirintą visomis kryptimis, tikrai prieisime prie išėjimo. Kam dar betaikyti kokį nors algoritmą? 

Mūsų uždavinys nuo klasikinio šiek tiek skiriasi: neprašoma rasti trumpiausio išėjimo iš labirinto, o prašoma tiesiog išeiti iš jo. Pats labirintas nėra duotas. Nepasakyta net, iš kurios pusės yra įėjimas į labirintą. 

Bangos ar paieškos į plotį algoritmą galima taikyti ir neapėjus viso labirinto. Tačiau jie nagrinėja ne gretimus labirinto langelius, o langelius, kurie vienodai nutolę nuo labirinto įėjimo. Reikia nepamiršti, kad šiame uždavinyje vaikštoma labirintu ir norint pakliūti nuo vieno tokio (vienodai nutolusio) langelio prie kito, gali tekti sukarti nemažą atstumą.

Paprasčiausia taikyti grįžimo metodą. Šiuo atveju jį galima pavadinti Kairiosios rankos taisykle – visą laiką einame laikydamiesi kairiosios sienos ir išėjimą būtinai pasieksime. Lygiai taip pat galime pasirinkti ir dešiniąją sieną ir eiti pagal ją. Taip vaikštant nereikia įsiminti nei ėjimų, nei paties labirinto. 

Pradžioje reikia nustatyti, iš kurios pusės yra įėjimas į labirintą. Tai nesunku padaryti: reikia kreiptis į funkcijas šiaurė, pietūs, rytai, vakarai. Bus leidžiama eiti tik viena iš šių krypčių. Mat sąlygoje pasakyta, kad įėjimą iš abiejų pusių riboja sienos. Taigi akivaizdu, kad į dvi puses negalėsime eiti. Grįžti pro įėjimą taip pat negalima. Šitai nepasakyta sąlygoje, tačiau tai galima nustatyti išbandžius modulį. 

Tarkime, kad eiti leidžiama į šiaurę. Tai reiškia, kad eidami pirmyn eisime į šiaurę. Jei beeidami labirintu nuspręsime, kad norime eiti į vakarus, teks eiti kairėn. Kai kreipiamės į procedūrą kairėn, prieš paeidami pasisukame į kairę. T. y. atlikę šį žingsnį, žiūrime vakarų kryptimi. Vaikštant labirintu svarbu sekti, kuria kryptimi žiūrime, t. y. kokia kryptimi (šiaurės, pietų, rytų, vakarų) eisime darydami žingsnį pirmyn. 

Taigi iš labirinto išeinama šitaip: jei galima, einama į kairę, jei ne, sužinoma, ar galima eiti pirmyn. Jei galima, ten einama, jei ne – bandoma eiti dešinėn, jei ir ten negalima – tada einama atgal.

Prieš pateikiant sprendimą reikia pakomentuoti porą dalykų. Modulyje pateikta funkcija dydis. Ji sąlygoje buvo pateikta sąmoningai, nors geram uždavinio sprendimui ji visiškai nereikalinga. Antra, sąlygoje prašoma, kad ėjimų skaičius labirinte būtų kuo mažesnis. Tai užuomina į sprendimą: jei bandysite taikyti bangos ar paieškos į plotį algoritmą, rasite trumpiausią kelią, bet ėjimų padarysite labai daug...

program labirintas;

  uses labir;

  type kryptys = (kair, pirm, deš, atg);

       šalys = (vak, šiaur, ryt, piet);



  function pred_š (šalis: šalys): šalys;

  begin

    if šalis = vak

       then pred_š := piet

       else pred_š := pred (šalis);

  end; { pred_š }



  function succ_š (šalis: šalys): šalys;

  begin

    if šalis = piet

       then succ_š := vak

       else succ_š := succ (šalis)

  end; { succ_š }



  function galima (kryptis: kryptys; šalis: šalys)

                                     : boolean;

  { nustato, ar galima eiti nurodyta kryptimi, jei žmogus veidu }

  { atsisukęs kryptimi šalis }

  begin

    case kryptis of

      kair: šalis := pred_š (šalis);

      deš: šalis := succ_š (šalis);

      atg: šalis := pred_š (pred_š (šalis))

    end;

    case šalis of

      vak: galima := vakarai;

      šiaur: galima := šiaurė;

      piet: galima := pietus;

      ryt: galima := rytai

    end;

  end; { galima }



  procedure eiti (kryptis: kryptys; var šalis: šalys);

  { atliekamas ėjimas nurodyta kryptimi }

  { rezultatas – pasaulio šalis, į kurią lieka atsisukęs žmogus po ėjimo }

  begin

    case kryptis of

      kair: kairėn;

      deš: dešinėn;

      atg: atgal;

      pirm: pirmyn;

    end;

    case kryptis of

      kair: šalis := pred_š (šalis);

      des: šalis := succ_š (šalis);

      atg: šalis := pred_š (pred_š (šalis))

    end

  end; { eiti }



  var kryptis: kryptys;

      paeita: boolean;

      šalis: šalys;

begin

  { nustatoma kuria kryptimi žiūri žmogus stovintis labirinto }

  { įėjime }

  if šiaurė

     then šalis := šiaur

     else if pietūs

             then šalis := piet

             else if rytai

                     then šalis := ryt

                     else šalis := vak;

  { kol neišėjome iš labirinto }

  while true do

    begin

      kryptis := kair; { bandysime eiti į kairę }

      paeita := false;

      while not paeita do

        if galima (kryptis, šalis) { jei galima paeiti šia kryptimi }

           then begin

                  eiti (kryptis, šalis);

                  paeita := true

                end

           else kryptis := succ (kryptis)

    end

end.



Olimpiados dalyviams modulio labir tekstas nebuvo pateiktas, tik sukompiliuotas modulis. Norintys panagrinėti modulio tekstą, gali jį rasti diskelyje. Diskelyje galite rasti ir kitą modulį lab_dem. Šis modulis piešia ekrane labirintą ir demonstruoja kaip jame atliekami jūsų ėjimai. Jį galite naudoti, jei labirintas nedidelis, t. y. dar telpa ekrane.



Testo nr.�

Labirinto �Ėjimų skaičius, jei einama �

Paaiškinimai���dydis�pagal kairįjį šoną�pagal dešinįjį šoną���1�10�37�37�Labirintas, kuriame nėra ciklų;��2�20�68�74�Labirintas, kuriame lengva užsiciklinti;��3�10�15�17�Labirinto vidus tuščias – jame nėra sienų;��4�10�18�14�Įėjimas į labirintą iš rytinės pusės;��5�10�15�17�Įėjimas į labirintą iš vakarinės pusės;��6�10�28�4�Įėjimas į labirintą iš pietinės pusės;��7�10�5�6�Labirintas turi daug išėjimų;��8�100�9411�293�“Šukų” formos labirintas��9�100�292�9410�“Šukų” formos labirintas; Šis testas simetriškas 8 testui;��10�100�195�197�“Dvigubų šukų” formos labirintas;��

136. Nameliai. Pirmas algoritmo žingsnis turbūt akivaizdus visiems: pagaliukai surikiuojami jų ilgių didėjimo (ar mažėjimo) tvarka. Tuomet atrenkami namelių sienoms tinkami pagaliukai, t. y. vienodo ilgio pagaliukų ketvertai ir sunumeruojami. „Sienos numeris“ skamba labai keistai, todėl vartosime terminą „namelio numeris“.

Tada taikome grįžimo metodą – perrenkame visus įmanomus variantus. Rekursijos parametras – namelio numeris nr. Didžiausias namelių skaičius randamas taip:

Suskaičiuojame kiek daugiausia namelių galime sukonstruoti, jei nestatysime namelio su numeriu nr. Taip pat suskaičiuojame kiek namelių galime pastatyti, jei šį namelį statysime. Skaičiuojant šį atvejį būtina išbandomi visi jau pastatytų namelių stogams nesunaudoti pagaliukai, t. y. visi galimi namelio stogo variantai.



const MAX_P = 500; { maksimalus pagaliukų skaičius }

type  pagaliukai = array [1..MAX_P] of integer;

        lentelė = array [1..MAX_P] of boolean;



procedure rikiuoti (n: integer; var p: pagaliukai);

{ surikiuoja pagaliukus ilgių mažėjimo tvarka }

{ rikiuojama burbuliuko metodu }

   var i, a: integer;

begin

  i := 1;

  while i < n do

    if p[i] <= p[i + 1]

       then i := i + 1

       else begin  { sukeičiame i-tąjį ir (i+1)-mąjį elementą }

              a := p[i];

              p[i] := p[i + 1];

              p[i + 1] := a;

              if i > 1 then i := i - 1

                       else i := i + 1;

            end

end; { rikiuoti }



function min (x, y: integer): integer;

begin

  if x < y

     then min := x

     else min := y;

end; { min }



function max (x, y: integer): integer;

begin

  if x > y

     then max := x

     else max := y;

end; { max }



function trik (a, b, c: integer): boolean;

{ ar galima iš atkarpų a, b, c sudaryti trikampį }

begin

  trik := ((a + b) > c) and ((a + c) > b) and 

          ((b + c) > a)

end; { trik }



function namel (n: integer; p: pagaliukai): integer;

  var sien, nam, daug, i: integer;

      s: pagaliukai;

      nau: lentelė;



  function konstr (nr, kiek: integer): integer;

  { rekursinė funkcija, konstruojanti namelius }

    var viso, pag, i, j: integer;

        laisva, ieškoti: boolean;

  begin

    if nr > sien

    then begin

           if kiek > nam

              then nam := kiek;

           konstr := kiek;

         end

    else begin

           { galime šio namelio nekonstruoti }

           viso := konstr(nr+1, kiek);

           { galime konstruoti namelį su numeriu nr }

           { tikriname, ar nesunaudoti sienoms skirti pagaliukai }

           laisva := true;

           pag := s[nr];

           for i := 0 to 3 do

             if nau[pag + i]

                then laisva := false;

           if not laisva { jei pagaliukai sunaudoti }

              then begin { daugiau nieko negalime padaryti }

                     konstr := viso;

                     exit;

                   end;

           { sunaudosime sienoms skirtus pagaliukus }

           for i := 0 to 3 do

             nau[pag + i] := true;

           { rasime stogą nameliui su numeriu nr }

           ieškoti := (nam < daug) and 

                      ((kiek + sien-nr+1) > nam);

           { jei neradome maksimalaus skaičiaus namelių ir }

           { galime rasti daugiau nei turime }

           i := 1;

           while (i <= n) and ieškoti do

             begin

               j := i + 1;

               while (j <= n) and ieškoti do

                 begin

                   if not nau[i] and not nau[j] and

                      trik (p[i], p[j], p[s[nr]])

                   then begin { radome stogą }

                         nau[i] := true; 

                         nau[j] := true;

                         viso := 

                           max(viso,konstr(nr+1,kiek+1));

                         nau[i] := false; 

                         nau[j] := false;

                         ieškoti := (nam < daug) and

                               ((kiek + sien-nr) > nam);

                        end;

                   { praleidžiami vienodo ilgio pagaliukai }

                   while (j <= n-1) and (p[j] = p[j+1]) do

                      j := j + 1;

                   j := j + 1;

                   if p[i] + p[s[nr]] <= p[j]

                   { jei j-asis pagaliukas jau per ilgas }

                      then j := n + 1;

                 end;

               i := i + 1;

             end;

           { atstatome buvusias reikšmes }

           for i := 0 to 3 do

             nau[pag + i] := false;

           konstr := viso

         end;

  end; { konstr }



begin

  { pagaliukai surikiuojami ilgių didėjimo tvarka }

  rikiuoti (n, p);

  { surasime visus galimas namelių sienas }

  sien := 0;

  for i := 1 to n-3 do

    if p[i] = p[i+3]

       then begin { radome naują sieną }

              sien := sien + 1;

              s[sien] := i; { įsimename vietą }

            end;

  daug := min (sien, n div 6); { daugiau namelių sudaryti negalima }

  nam := 0; { didžiausias sukonstruotų namelių skaičius }

  { nei vienas pagaliukas dar nepanaudotas }

  for i := 1 to n do

    nau[i] := false;

  if daug = 0

     then namel := 0

     else namel := konstr(1, 0);

end; { namel }



Manantiems, kad uždavinio sprendimui tinka Šykštusis algoritmas siūlome panagrinėti pavyzdį, kai duota 18 pagaliukų: 2, 3, 3, 4, 5, 5, 5, 5, 50, 53, 53, 53, 53, 500, 500, 500, 500, 500.

Iš jų galima sudaryti tris namelius (136.2 pav.):

136.2 pav.



Testo nr.�Pagaliukų skaičius�Rezultatas�Paaiškinimas��1�7�0�Nedidelis testas��2�31�5�Yra 6 blokai namelių sienoms��3�31�6�Yra 7 blokai namelių sienoms��4�31�5�Yra 8 blokai namelių sienoms��5�31�4�Yra 10 blokai namelių sienoms��6�18�3�Testas sutampa su aukščiau pateiktu pavyzdžiu��

Vyresniems (X – XII kl.)

137. Siųstuvai ir imtuvas. Šis uždavinys yra atskiras 129 uždavinio variantas. Iš pirmo žvilgsnio jis atrodo žymiai paprastesnis: siųstuvų jungimo tvarka aiški, tereikia perrinkti visus įmanomus variantus, reikia tiktai supaprastinti 129 uždavinio algoritmą.

Siųstuvus jungsime nuo pirmojo iki n-ojo. Sakykime, kad sėkmingai įjungėme pirmuosius k siųstuvų. k-asis siųstuvas buvo įjungtas momentu laikask .Paskutinysis šių siųstuvų signalas imtuvą pasieks laiko momentu pask_e. Taigi, jungiant (k+1)-ajį siųstuvą jį reikia bandyti jungti visais laiko momentais nuo laikask+1 iki pask_e+1. Deja, bandydami jungti visais šiais momentais, gauname labai daug panašių variantų ir vykdymo laikas labai pailgėja.

Tenka ieškoti kito algoritmo. Siųstuvus jungsime atvirkščiai... – nuo n-ojo iki pirmojo. Pradėsime nuo n-ojo siųstuvo, kurį įjungsime nuliniu laiko momentu. Paskutinis šio siųstuvo signalas pasieks imtuvą momentu ilgn, (čia ilgn – šio siųstuvo pranešimo perdavimo laikas). Taigi optimaliame sprendinyje nuo n-tojo siųstuvo įjungimo laiko iki momento, kada paskutinysis signalas pasieks imtuvą gali praeiti ilgn arba daugiau sekundžių, tačiau negali praeiti mažiau sekundžių. Šią siųstuvo charakteristiką (minimalų laiką nuo siųstuvo įjungimo iki momento kada paskutinysis signalas pasieks imtuvą) pavadinkime opti, kur i – siųstuvo numeris.

Pirmą kartą jungėme tik n-ąjį siųstuvą, antrą kartą jungiame jau (n-1)-ąjį ir n-ąjį ir t. t. Kiekvieną kartą pradedame jungti nuo vienu numeriu žemesnio siųstuvo ir radę optimalų šių siųstuvų jungimo variantą kartu randame ir to siųstuvo charakteristiką opti. 

Sakykime, kad mokame optimaliai įjungti siųstuvus nuo i-ojo iki n-ojo. Ieškosime kaip optimaliai sujungti siųstuvus su numeriais i-1, i, i+1, ...n. Pažymėkime pask laiko momentą, kada paskutinysis šių siųstuvų signalas pasiekia imtuvą kuriame nors teisingame, tačiau nebūtinai optimaliame sprendinyje. Tokį sprendinį rasti nesunku: galima siųstuvus jungti kas 200 sekundžių.

(i-1)-ąjį siųstuvą jungiame nuliniu laiko momentu: laikasi-1 ( 0;

i-tąjį siųstuvą bandome jungti visais laiko momentais nuo laikasi-1+1 iki pask - opti. 

Tai yra laikasi ( [ laikasi-1+1; pask - opti]. Jei žinome, kad nuo i-ojo siųstuvo įjungimo iki darbo pabaigos būtinai turi praeiti opti sekundžių, tai nėra prasmės nagrinėti atvejų, kur laikasi + opti ( pask;

Analogiškai (i+1)-ąjį siųstuvą bandome jungti laiko momentais �laikasi+1 ( [ laikasi+1; pask - opti+1]

const MAX_N = 20; { maksimalus siųstuvų skaičius }

      LAIK = 200; { pranešimo perdavimo laikas }

type siųstuv = array [1..MAX_N] of integer;

     siųstuv_l = array [1..MAX_N] of boolean;

     lentelė = array [1..MAX_N, 1..LAIK] of integer;

     imtuvas = array [1..LAIK*MAX_N] of boolean;



{ žemiau surašyti globalieji kintamieji }

var n: integer; { siųstuvų skaičius }

    t, l, p: siųstuv; { siųstuvų charakteristikos }



    skl: lentelė; { rodo kada imtuvą pasiekia kiekvieno }

    { siųstuvo signalai, jei siųstuvas įjungiamas nuliniu momentu }



procedure skleistinė (var skl: lentelė);

{ kada imtuvą pasiekia kiekvieno siųstuvo signalai }

{ jei siųstuvas įjungiamas nuliniu laiko momentu }

  var k, sek, sign: integer;

begin

  for k := 1 to n do

    for sign := 1 to LAIK do

      skl[k, sign] := 0;

  for k := 1 to n do

    begin

      skl[k, 1] := l[k];

      sek := l[k];

      for sign := 2 to p[k] do

        begin

          sek := sek + t[k];

          skl[k, sign] := sek;

       end

    end;

end; { skleistinė }



procedure įjungti (k, sek: integer;

                   var imt: imtuvas; var galima: boolean);

{ k-ąjį siųstuvą bando įjungti laiko momentu sek }

{ jei galima tai padaryti, tai pakeičiamas masyvas imt }

  var sign, i: integer;

begin

  sign := 1;

  galima := true;

  while (skl[k, sign] <> 0) and galima do

    begin

      if imt[sek + skl[k, sign]]  { jei vieta užimta }

         then begin { atstatome atgal }

                for i := sign-1 downto 1 do

                  imt[sek + skl[k, i]] := false;

                galima := false

             end

         else imt[sek + skl[k, sign]] := true;

      sign := sign + 1

    end;

end; { įjungti }



procedure išjungti (k, { siųstuvo numeris }

                    sek: integer; { kada jis buvo paleistas }

                    var imt: imtuvas);

{ išjungia k-ąjį siųstuvą (atstato masyvą imt) }

  var sign: integer;

begin

  sign := 1;

  while skl[k, sign] <> 0 do

    begin

      imt[sek + skl[k, sign]] := false;

      sign := sign + 1

    end;

end; { išjungti }



procedure jungimas (var kada: siųstuv);

{ randa optimalią siųstuvų įjungimo tvarką }

  var laikas, { kada pradeda veikti kiekvienas siųstuvas }

      ilg: siųstuv; { siųstuvų pranešimų perdavimo ilgiai }

      k, i,

      pask, { laiko momentas, kada imtuvą pasiekia }

            { paskutinis signalas optimaliame sprendinyje }

      pask_e: integer; { ir nagrinėjamame sprendinyje }

      imt: imtuvas;  { kuriais momentais imtuvą pasiekia signalai }

      gerai: boolean;

      opt: siųstuv;



  procedure dėlioti (k, pask_e: integer; laikas: siųstuv;

                     var kada: siųstuv; { optimali jungimo tvarka }

                     var pask: integer);

  { rekursinė procedūra, dėliojanti siųstuvus }

  { k – nagrinėjamas siųstuvas }

  { pask_e – kada imtuvą pasiekia jau įjungtų siųstuvų paskutinis signalas }

    var senas, sek: integer;

        gerai: boolean;

  begin

    if k > n     { jei visi siųstuvai įjungti }

    then begin   { tai radome geresnį sprendinį }

           kada := laikas;

           pask := pask_e;

        end

    else begin

           sek := laikas[k-1] + 1;

           senas := pask_e;

           while sek + opt[k] < pask do { jei yra prasmė jungti }

             begin

               { bandome jungti }

               įjungti (k, sek, imt, gerai);

               if gerai then

                  begin

                    laikas[k] := sek;

                    if pask_e < sek + ilg[k]

                       then pask_e := sek + ilg[k];

                    dėlioti (k+1, pask_e, laikas, kada, pask);

                    pask_e := senas;

                    išjungti (k, sek, imt);

                    if k = n then sek := pask;

                  end;

               sek := sek + 1;

             end;

         end;

  end; { dėlioti }



  var j: integer;

begin

  skleistinė (skl);

  { randame kiekvieno pranešimo ilgį }

  for k := 1 to n do

    ilg[k] := l[k] + (p[k]-1) * t[k];

  opt[n] := ilg[n];

  for i := n-1 downto 1 do

    { optimaliai sujungsime siųstuvus nuo i-ojo iki n-ojo }

    begin

      { imtuvo nepasiekė nei vienas signalas }

      for j := 1 to LAIK*MAX_N do

         imt[j] := false;

      { i-ąjį siųstuvą įjungiame nuliniu momentu }

      laikas[i] := 0;

      įjungti (i, 0, imt, gerai);

      pask_e := ilg[i];

      { randame teisingą, tačiau nebūtinai optimalų sprendinį }

      for k := i to n do

        begin

          if k = 1

             then kada[k] := 0

             else kada[k] := pask + 1;

          pask := kada[k] + ilg[k];

        end;

      dėlioti (i+1, pask_e, laikas, kada, pask);

      opt[i] :=  pask;

    end;

end; { jungimas }



Testo nr.�Pradiniai duomenys/

siųstuvų skaičius�Galimi rezultatai�Laiko momentas, kada paskutinis signalas pasiekia imtuvą�Paaiškinimai��1�3

5 1 2

2 4 3

3 2 2�0 1 6�11�Duoti tik trys siųstuvai; paprastas testas, kurio pagalba lengva ieškoti klaidų sprendime; tinka ir šykštusis algoritmas��2�4

2 2 3

2 2 3

2 2 3

2 2 3�0 1 6 7�13�Duoti keturi siųstuvai, jų visos charakteristikos sutampa; Tinka ir šykštusis algoritmas��3�4

...�0 18 40 41 �64�Tinka ir šykštusis algoritmas��4�3

3 2 4

2 3 4

2 2 4�0 7 11 �19�Tinka ir šykštusis algoritmas��5�6

...�0 1 9 74 164 206�291�Tinka ir šykštusis algoritmas, tikrinamas sprendimo efektyvumas��6�7

...�0 2 4 13 14 15 28�72�Tinka ir šykštusis algoritmas��7�3

5 25 10

5 15 10

6 25 17�0 4 26�147�Nedidelis testas, kuriam netinka šykštusis algoritmas; jo pagalba lengva ieškoti klaidų sprendime��8�15

...�0 7 15 20 23 24 32 37 50 52 57 69 71 83 89�103�Tikrinamas sprendimo efektyvumas��9�14

...�0 2 24 27 30 40 41 45 52 59 63 68 75 81�100�Tikrinamas sprendimo efektyvumas��10�10

...�0 23 25 26 60 108 110 113 119 120�282�Tikrinamas sprendimo efektyvumas��Pastaba. Rezultatai surašyti į vieną ar dvi eilutes, bet ne po vieną skaičių eilutėje kaip to reikalauja sąlygą.

138. Sukabinti žiedai. Sumodeliuokime situaciją naudodamiesi grafais. Grafo viršūnėmis laikykime žiedus. Viršūnės a ir b sujungtos briauna, jei žiedai a ir b sukabinti tarpusavyje. Gavome neorientuotą grafą. 

Imkime bet kurią vientisą, be išsišakojimų grandinę, jungiančią žiedus a ir b. Tai kelias grafe, jungiantis viršūnes a ir b ir neturintis ciklų. Jei kelio ilgis yra l, tai tame kelyje sutinkama l+1 viršūnė (žiedas). Jei iš viso turime n žiedų, galime pašalinti n - (l+1) žiedą.

138.2 pav. nubrėžtas grafas, atitinkantis pradinių duomenų pavyzdį sąlygoje:

138.2 pav.

Sąlygą prašo rasti kiek daugiausia žiedų galima pašalinti. Kuo mažiau žiedų bus grandinėje, jungiančioje žiedus a ir b, tuo daugiau žiedų bus galima pašalinti.

Randame trumpiausio kelio, jungiančio žiedus a ir b ilgį min_l. Grandinėje, atitinkančioje šį kelią yra min_l + 1 žiedas. Vadinasi, daugiausia galime pašalinti �n - min_l + 1 žiedą.

Trumpiausio kelio grafe tarp dviejų viršūnių galima ieškoti įvairiais būdais. Mes pasirinkome Paieškos į plotį algoritmą (žr. 83 užd).

Pats grafas uždavinyje vaizduojamas neįprastai – porų masyvų. Mat kvadratinė lentelė, aprašyta tokiu duomenų tipu:

type grafas ( array [1..300, 1..300] of boolean;

užimtų apie 88K, tuo tarpu, kai Turbo Paskalis statiniams duomenims leidžia išskirti ne daugiau 64K.

const MAX_N = 300;  { maksimalus žiedų skaičius }

      MAX_K = 1000; { maksimalus sukabinimų skaičius }

type  grafas = array [1..MAX_K] of record

                                     v1, v2: integer

                                   end;

      viršūnės = array [1..MAX_N] of  

                         record

                           sp: (balta, pilka);

                           d: integer { trumpiausias atstumas }

                         end;         { iki a }



      eilė = array [1..MAX_N] of integer;

             { į eilę įtraukiami tik viršūnių numeriai }



procedure įtraukti_į_eilę (x: integer;  var e: eilė; 

                           var uodega: integer);

{ įtraukia į eilės e galą viršūnę x }

begin

  e[uodega] := x;

  if uodega = MAX_N

     then uodega := 1

     else uodega := uodega + 1

end; { įtraukti_į_eilę }



procedure pašalinti_iš_eilės (var galva: integer);

{ pašalina iš eilės e pirmąją viršūnę }

begin

  if galva = MAX_N

     then galva := 1

     else galva := galva + 1

end; { pašalinti_iš_eilės }



function žiedai(a, b, k, n: integer; g: grafas): integer;

  var galva, uodega, { eilės atributai }

      u, v, i: integer;

      atr: viršūnės; { visų viršūnių atributai }

      e: eilė;       { viršūnių eilė }

begin

  { inicializuojame viršūnes }

  for i := 1 to n do

    begin

      atr[i].sp := balta;

      atr[i].d := maxint;

    end;

  atr[a].sp := pilka;  { žinome trumpiausią atstumą iki a }

  atr[a].d := 0;

  galva := 1;

  uodega := 1;

  įtraukti_į_eilę (a, e, uodega);

  while galva <> uodega do   { kol eilė netuščia }

    begin

      u := e[galva];

      for i := 1 to k do

        if (g[i].v1 = u) and (atr[g[i].v2].sp = balta) or

           (g[i].v2 = u) and (atr[g[i].v1].sp = balta)

           { randame dar nenagrinėtą viršūnę, gretimą viršūnei u }

           then begin

                  if g[i].v1 = u { imame reikiamą viršūnę iš poros }

                     then v := g[i].v2

                     else v := g[i].v1;

                  atr[v].sp := pilka;

                  atr[v].d := atr[u].d + 1;

                  įtraukti_į_eilę (v, e, uodega)

                end;

      pašalinti_iš_eilės (galva);

    end;

  { rasime pašalinamų žiedų skaičių }

  žiedai := n - (atr[b].d + 1)

end; { žiedai }





Testo nr.�Rezultatas��1�11��2�9��3�30��4�41��5�58��6�139��7�189��8�227��

139. Erdvėlaivis. Jei kreditų skaičius būtų neribojamas, tai būtų trumpiausio kelio paieškos grafe uždavinys, kurį būtų galima spręsti pagal Dijkstros algoritmą ar kokį kitą dar paprastesnį algoritmą. Šiuo atveju situacija sudėtingesnė.

Kokiu būdų bespręstume uždavinį, būtinai reikės tikrinti, ar užteks kuro, esančio bake iš vienos bazės nuskristi į kitą. Todėl pradžioje apskaičiuokime atstumus tarp visų bazių. Sakykime, kad turime dvi bazes a(x1, y1, z1) ir b(x2, y2, z2). Atstumas tarp jų skaičiuojamas pagal formulę:

d ( ((x1 - x2)2 + (y1 - y2)2 + (z1 - z2)2

Na o maksimalus atstumas, kurį galima nuskristi su pilnu kuro rezervuaru lygus v/san, čia v – kuro rezervuaro talpa, san – kuro sąnaudos vienam ilgio vienetui.

const MAX_N = 100; { maksimalus bazių skaičius }

type  bazės = array [1..MAX_N] of 

                             record { duomenys apie bazes }

                               k, x, y, z: integer;

                             end;

        atstumai = array [1..MAX_N, 1..MAX_N] of real;

                   { atstumai tarp bazių }



procedure atstum (n: integer; baz: bazės;

                  var ats: atstumai);

{ suskaičiuoja atstumus tarp bazių }

  var i, j: integer;

      d: real;

begin

  for i := 1 to n do

    for j := 1 to n do

      ats[i, j] := 0;

  for i := 1 to n do

    for j := i+1 to n do

      begin

        d := sqrt(sqr(baz[i].x - baz[j].x) +

                  sqr(baz[i].y - baz[j].y) +

                  sqr(baz[i].z - baz[j].z));

        ats[i, j] := d;

        ats[j, i] := d;

    end;

end; { atstum }



Uždavinį spręsime Dinaminio programavimo metodu. Susidarysime dvimatę lentelę, turinčią tiek eilučių kiek yra bazių ir tiek stulpelių, kiek yra kreditų. Lentelę pavadinkime m. Kiekvienas lentelės elementas susidės iš dviejų laukų. m[a, kr].kur rodys bazę į kurią reikia skristi iš bazės a, jei turime kr kreditų. Jei turint kr kreditų iš bazės a neįmanoma pasiekti galutinės bazės, tai m[a, kr].kur ( -1. m[a, kr].ats_n yra trumpiausio kelio ilgis nuo bazės a iki galutinės bazės, jei turime k kreditų. 

Nesunkiai galime užpildyti pirmąjį lentelės stulpelį ir paskutiniąją eilutę. Mat m[a, 0].kur ( -1, nebent nereikėtų mokėti už kurą bazėje a ir jo užtektų kuro be sustojimų pasiekti bazę n. Taip pat akivaizdu kad m[N, kr].kur ( 0, ir m[N, kr].ats_n ( 0. Mat iš paskutiniosios bazės niekur skristi nereikia. 

Imkime sąlygoje patiktą pavyzdį ir pradėkime pildyti lentelę. Su pilnu baku kuro galima nuskristi 10 ilgio vienetų.

Atstumai tarp bazių

bazė�1�2�3�4��1�0.00�10.00�7.07�10.20��2�10.00�0.00�7.07�2.00��3�7.07�7.07�0.00�5.83��4�10.20�2.00�5.83�0.00��

kreditas

/bazė�0�1�2�3�4�5�6�7�8�9�10��1�-1�������������������������2�-1�������������������������3�-1�������������������������4�0�0�0�0�0�0�0�0�0�0�0���0.00�0.00�0.00�0.00�0.00�0.00�0.00�0.00�0.00�0.00�0.00��

Lentelę pildysime iš kairės į dešinę. Pirmasis lentelės stulpelis jau užpildytas. Sakysime, kad turime k kreditų, o kuro bako papildymas bazėje a kainuoja kred kreditų. Išskridus iš bazės a liks k - kred kreditų. Peržiūrime visas bazes iki kurių be sustojimo galime nuskristi iš a. Iš jų išsirenkame tas, iš kurių turėdami k - kred kreditų galėsime pasiekti galutinę bazę. T. y. peržiūrėdami reikiamus anksčiau užpildytus stulpelius, pildome tolesnį stulpelį. 

Nagrinėjamam pavyzdžiui užpildyta lentelė atrodytų taip:

kreditas

/bazė�0�1�2�3�4�5�6�7�8�9�10��1�-1�-1�-1�-1�-1�-1�-1�-1�-1�-1�3�������������12.90��2�-1�-1�-1�-1�-1�-1�-1�-1�-1�-1�-1���������������3�-1�-1�-1�-1�-1�4�4�4�4�4�4��������5.83�5.83�5.83�5.83�5.83�5.83��4�0�0�0�0�0�0�0�0�0�0�0���0.00�0.00�0.00�0.00�0.00�0.00�0.00�0.00�0.00�0.00�0.00��

Keblumų sukelia atvejis, kai bazėje už kurą mokėti nereikia. Tokiu atveju �k - kred = k. Šiuo atveju tektų pildyti k-tąjį stulpelį, ir nagrinėti to paties stulpelio dar neužpildytus langelius m[b, k].kur. Problema sprendžiama taip. Pildant stulpelį, pirmiausia užpildomi tų bazių, kuriose reikia mokėti už kurą langeliai. Tuomet pildomi bazių, kuriose mokėti nereikia, langeliai. Jei bazėje b į kurią skrendame už kurą mokėti reikia, tai naudojame jau užpildytą langelio m[b, k].kur reikšmę. Jei bazėje b už kurą mokėti nereikia, imame langelio m[b, k-1].kur reikšmę. 

Užrašant algoritmą programa iškyla atmintis problema. Mat aukščiau minėtai lentelei trūktų atminties. Todėl lentelę suskaidome į dvi lenteles: bazių į kurias reikia skristi ir atstumų lentelę. Pirmajai vartojame statinę, antrajai – dinaminę atmintį.



const MAX_N = 100; { maksimalus bazių skaičius }

      MAX_K = 100; { maksimalus kreditas }

      DAUG = 1000; { atstumas tarp dviejų bazių nebus didesnis už šį }

type  bazės = array [1..MAX_N] of record { duomenys apie bazes }

                                    k, x, y, z: integer;

                                  end;

      atstumai = array [1..MAX_N, 1..MAX_N] of real;

                      { atstumai tarp bazių }

      maršrutas = array [1..MAX_N, 0..MAX_K] of integer;

   { elementas [a, kr] parodo, į kurią bazę reikia skristi iš bazės a, turint kreditą kr }

      t_atst = array[1..MAX_N, 0..MAX_K] of real;

      atstumas_n = ^t_atst;

      { parodo, koks trumpiausias atstumas iki n-tosios bazės }

      { jei esame bazėje a ir turime kr kreditų }



procedure skrieti (v, san,{ rezervuaro talpa ir kuro sąnaudos ilgio vnt. }

                   kr, n: integer; 

                   { turimas kreditas ir bazių skaičius }

                   baz: bazės; var kur: maršrutas);

{ randa optimalų skridimo maršrutą }

  var ats: atstumai;

      min,

      max_kel: real;   { ilgiausias kelias, kurį galima }

                       { nuskristi su pilnu baku kuro }

      ats_n: atstumas_n; { trumpiausi atstumai iki n-tosios bazės }

      i, k, a, b, lik: integer;



  procedure inicializuoti (var kur: maršrutas; 

                           var ats_n: atstumas_n);

  { užpildo lentelių pirmąjį stulpelį ir paskutinę eilutę }

  { vartoja globalius kintamuosius n, kr, max_kel, ats, baz }

  { šių kintamųjų reikšmių nekeičia }

    var a, k: integer;

  begin

    new (ats_n);

    { užpildysime kur[a, 0] }

    for a := 1 to n-1 do

      if (baz[a].k = 0) and (max_kel >= ats[a, n])

         then begin

                kur[a, 0] := n;

                ats_n^[a, 0] := ats[a, n];

              end

         else kur[a, 0] := -1; { nuskristi be mokesčio negalima }

    { užpildysime kur[n, k] }

    for k := 0 to kr do

      begin

        kur[n, k] := 0; { toliau skristi nebereikia }

        ats_n^[n, k] := 0;

      end;

  end; { inicializuoti }



  procedure skristi_a (a, k: integer; { skrenda iš a su kreditu k }

                       var kur: maršrutas; 

                       var ats_n: atstumas_n);

  { randa trumpiausią kelią iš bazės a į bazę n, kai turima k kreditų }

  { vartojami globalieji kintamieji max_kel, baz, ats }

  { šių kintamųjų reikšmės nekeičiamos }

  var min: real; { minimalus atstumas nuo bazės a iki bazės n }

      b, 

      lik: integer; { kredito likutis }

  begin

    min := DAUG; 

    kur[a, k] := -1; { kelio kol kas neradome }

    for b := 1 to n do { skrendame į bazę b }

      if (ats[a, b] <= max_kel) and (a <> b) 

         { jei bazė b pasiekiama }

      then begin

             if (baz[a].k = 0) and (baz[b].k = 0)

                then lik := k - 1

                else lik := k - baz[a].k;

             if (kur[b, lik] >= 0) 

                { jei iš bazės b pasieksim n-tąją bazę }

                and (min > (ats[a, b] + ats_n^[b, lik]))

                { ir naujas kelias yra trumpesnis }

                then begin

                       min := ats[a, b] + ats_n^[b, lik];

                       kur[a, k] := b;

                       ats_n^[a, k] := min;

                     end;

           end;

  end; { skristi }



begin

  { rasime atstumus tarp bazių }

  atstum (n, baz, ats);

  max_kel := v/san;

  inicializuoti (kur, ats_n);

  { užpildysime likusią lentelės dalį }

  for k := 1 to kr do

    begin { turime k kreditų }

      { pirmiausia skrendame iš bazių, kuriose reikia mokėti }

      for a := 1 to n-1 do { skrendame iš bazės a }

        if k < baz[a].k { jei iš šios bazės negalima išskristi }

          then kur[a, k] := -1

          else if baz[a].k <> 0

                  then skristi_a (a, k, kur, ats_n);

      { tuomet skrendame iš bazių, kuriose nereikia mokėti }

      for a := 1 to n-1 do

        if baz[a].k = 0

           then skristi_a (a, k, kur, ats_n);

    end;

end; { skrieti }



Testo nr.�Rezultatai 

(surašyti vienoje eilutėje)�Komentarai��0�1 3 4�Testas sutampa su sąlygoje pateiktu pavyzdžiu��1�1 9 12 15 4 20�Nedidelis testas, kuriame nėra nemokamų bazių��2�1 97 28 58 100�Didelis testas, kuriame yra nemokamų bazių (tikrina algoritmo efektyvumą)��3�1 2 3 ...  94 95�Visos bazės yra vienoje tiesėje, atstumai tarp jų vienodi, kuro užtenka nuskristi tik į dvi gretimas bazes��4�1 49 7 16 53 73 79 96 100�Didelis testas, kuriame nėra nemokamų bazių (tikrina algoritmo efektyvumą)��5�1 8 9 10�Nei vienoje bazėje už kurą nereikia mokėti��

140. Pokeris. Uždavinys jungia kombinatorikos ir tikimybių teorijos elementus. Sprendimas gana ilgas, todėl jį pateiksime dalimis ir kiekvieną dalį nagrinėsime atskirai. 

Sąlyga prašo surasti tokią paliekamų ir keičiamų kortų kombinaciją, kuriai esant išlošimo vidurkis būtų didžiausias. Pradžioje laikome, kad išlošimo vidurkis lygus nuliui. Tuomet konstruojame visas įmanomas paliekamų ir keičiamų kortų kombinacijas ir skaičiuojame kiekvienos kombinacijos išlošimo vidurkį (kaip jį skaičiuoti panagrinėsime vėliau). Kortų kombinacija, kurios išlošimo vidurkis didžiausias ir bus ieškomas rezultatas.

Žemiau pateikiame svarbiausią procedūrą spręsk:

const MAX_K = 5;  { duotų kortų skaičius }

      MAX_P = 8;  { išlošiančių padėčių skaičius }

      MAX_R = 6;  { kortų rangų skaičius }

type  korta = record

                ran: longint; { kortos rangas }

                sp : char     { kortos rūšis }

              end;

      kortos = array [1..MAX_K] of korta;

      išlošimas = array [1..MAX_P] of longint; 

                  { išlošiamos sumos }

      log_mas = array [1..MAX_K] of boolean;

                { paliekama korta ar keičiama }

      rangai = array [1..MAX_R] of longint;



procedure spręsk (k: kortos;      { duotosios kortos }

                  išl: išlošimas; { kiekvienos išlošiančios padėties }

                                  { išlošimo sumos }

                  var opt: log_mas 

              { uždavinio rezultatas – paliekamų ir keičiamų kortų sąrašas });

    var pal: log_mas;

        max_nauda: real; { maksimalus išlošimo vidurkis }



  procedure palikti_keisti (n: longint; pal: log_mas);

  { perrenkami visi įmanomi paliekamų ir keičiamų kortų deriniai }

  { naudojami globalūs kintamieji k, išl, max_nauda ir opt }

  { kintamųjų k ir išl reikšmės nekeičiamos }

  { procedūros rezultatas – kintamojo opt reikšmė }

    var nauda: real;

  begin

    if n = 6 { visos kortos arba paliktos arba nurodytos keisti }

       then begin

             nauda := kokia_nauda (k, išl, pal);

             if nauda > max_nauda

             then begin { rastas naudingesnis kortų pasilikimo variantas }

                    max_nauda := nauda;

                    opt := pal;

                  end

            end

       else begin

              pal[n] := false;

              palikti_keisti (n + 1, pal);

              pal[n] := true;

              palikti_keisti (n + 1, pal);

           end

  end; { palikti_keisti }

begin { spręsk }

  max_nauda := -1;

  palikti_keisti (1, pal);

end; { spręsk }



Dabar panagrinėsime kaip rasti išlošimo vidurkį, kai žinoma kurios kortos yra paliekamos. Tam reikia suskaičiuoti keliais skirtingais būdais galima sudaryti kiekvieną išlošiamą padėtį bei rasti visų galimų skirtingų kombinacijų skaičių. 

Išlošiamas kombinacijas suskirstome į du tipus:

a) Pora, Dvi poros, Trejetas, Pilni namai, Ketvertas.

b) Iš eilės, Vienos spalvos, Pokeris (šias kombinacijas galima sudaryti tik tada, kai visos paliktos kortos yra skirtingo rango).



a) Pora, Dvi poros, Trejetas, Pilni namai, Ketvertas.

Pradėsime nuo šio tipo. Visas įmanomas kombinacijas sudarinėsime rekursiškai. Žinome, kad mums reikia parinkti liko kortų bei kiek kiekvieno rango kortų yra palikta. Sudarykime lentelę vien, turinčią 4 elementus. Ji rodys, kiek tarp paliktų kortų yra Porų, Trejetų, Ketvertų.

Rekursiją vykdysime pagal kortų rangus. Devynakių (arba kortų, kurių rangas yra 1) galime parinkti bet kokį skaičių iš intervalo [0, min(liko, p9)], čia p9 yra paliktų devynakių skaičius. Pavyzdžiui, turėjome tokias kortas (140.1 pav.):

140.1 pav.

Nusprendėme palikti kryžių devynakę ir vynų valetą, o keisti vynų devynakę, vynų dešimtakę, būgnų tūzą.

Tarp paliktųjų kortų yra viena devynakė. Taigi parinkdami kortas, galime neparinkti nei vienos devynakės, parinkti, vieną, dvi ar tris. Keturių devynakių parinkti negalime todėl, kad reikia parinkti tik tris kortas ir todėl, kad viena devynakė jau parinkta. Taigi, devynakių skaičių galime parinkti tokiais būdais (140.2 pav.):

140.2 pav.



Parinkdami Devynakes pildome ir lentelę vien. 

Po to sumažiname kortų, kurias reikia parinkti, skaičių ir bandome parinkti Dešimtakes ir t. t. Kai parenkame visų rangų kortas, patikriname, ar turimų kortų skaičius lygus 5 ir suradę reikiamą išlošiamą padėtį padidiname tos padėties kombinacijų skaičių bei bendrą kombinacijų skaičių. Bendras kombinacijų skaičius yra didinamas net ir tada, jei jokia išlošiama padėtis nesusidaro. 

Atskirai reiktų pakalbėti apie tai, kaip skaičiuojamos kombinacijos. Imkime tą patį pavyzdį, kai pasilikome kortas kryžių devynakę ir vynų valetą. Šią kombinaciją gavome vieninteliu būdu: pasilikdami tas dvi kortas. Rinkdami kortas pagal rangą, pirmiausia renkame devynakes. Sakykime, kad parinkome dvi devynakes. Galime gauti tokias kombinacijas (140.3 pav.):

140.3 pav.

Galime pasakyti, taip: turime n (šiuo atveju trys – širdžių devynakė, būgnų devynakė, vynų devynakė) skirtingų kortų aibę. Iš jos reikia išrinkti poaibį iš k (šiuo atveju dvi) skirtingų kortų (t. y. negalime parinkti dviejų vynų devynakių). Toks poaibis vadinamas Deriniu be pasikartojimų iš n po k elementų. Skirtingų derinių be pasikartojimų skaičius lygus

                   n!

Cnk  (  ____________

              k! (n - k)!

Mūsų atveju C32  ( 3!/(2! (3-1)!) = 3.

Sakykime, kad parinkę du devynetus, nusprendėme parinkti dar vieną dešimtuką. Dešimtuką galime parinkti C41  ( 4!/(1!(3!) ( 4 būdais. Kombinaciją kryžių devynakė, vynų valetas, dvi devynakės ir viena bet kokia korta, galėjome gauti 3 būdais. Taigi, kombinaciją kryžių devynakė, vynų valetas, dvi devynakės ir dešimtakė galime gauti 3(4 ( 12 būdų (140.4 pav.):

140.4 pav.



function cNpok (n, k: longint): longint;

{ randa derinių be pasikartojimų skaičių: c iš n po k }

  var sk: longint;

      i: longint;

begin

   sk := 1;

   for i := n downto n-k+1 do sk := sk*i;

   for i := 2 to k do sk := sk div i;

   cNpok := sk

end; { cNpok }



function min (a, b: longint): longint;

begin

  if a < b then min := a

           else min := b

end; { min }



procedure perrinkti (ran, liko: longint; komb: longint);

{ randa Porų, Trejetų, Ketvertų kombinacijų skaičių, }

{ rekursijos gylį nurodo rangas ran, einama padėtis gali susidaryti komb būdais }

{ taip pat randa visų įmanomų kombinacijų skaičių }

{ procedūra naudoja globalius kintamuosius: vien, išviso, kiek }

{ procedūros rezultatai – išviso ir kiek reikšmės }

  var i: longint;

begin

  if ran > MAX_R { perbėgti visi rangai }

  then if liko = 0 { visos 5 kortos parinktos }

       then begin

         if vien[4] > 0 then kiek[7] := kiek[7] + komb

         else if vien[3] > 0

              then if vien[2] > 0

                   then kiek[6] := kiek[6]+komb

                   else kiek[3] := kiek[3]+komb

              else if vien[2] > 0

                   then kiek[vien[2]] := kiek[vien[2]] +

                                         komb;

              išviso := išviso + komb

         end

       else

  else for i := 0 to min (liko, 4-rng[ran]) do

       { imamas visoks įmanomas skaičius ran rango kortų }

         begin

          if i + rng[ran] > 0

          then vien[i+rng[ran]] := vien[i+rng[ran]] + 1;

          perrinkti(ran+1, liko-i, komb*cNpok (4-rng[ran], i));

          if i + rng[ran] > 0

          then vien[i+rng[ran]] := vien[i+rng[ran]] - 1;

        end;

end; { perrinkti }



b) Iš eilės, Vienos spalvos, Pokeris (šias kombinacijas galima sudaryti tik tada, kai visos paliktos kortos yra skirtingo rango). Patikrinę, ar turimos kortos yra skirtingo rango, imkimės kombinacijų skaičiavimo. 

Pradžioje tarkime, kad turime, tik vienos rūšies kortas (pavyzdžiui, kryžius). Žinome, kad yra MAX_R skirtingų kortų rangų, ir reikia sudaryti kombinaciją iš MAX_K kortų. Skaičiuodami kombinacijas kartu nagrinėsime pavyzdį, kai Pavyzdžiui, MAX_R ( 8 (prisideda septynakė ir aštuonakė) o reikia parinkti 5 kortas. Skirsime du atvejus:

- nėra palikta nei vienos kortos. tuomet, galime sudaryti �MAX_R - MAX_K + 1 skirtingą kombinaciją.

Mūsų pavyzdžiu tai galima padaryti 8-5+1 ( 4 būdais (140.5 pav.)

140.5 pav.

- yra palikta keletas kortų. Tuomet randame mažiausio ir didžiausio rango paliktas kortas. Visas tarp šių rangų esančias kortas, kurios nėra parinktos parenkame vienareikšmiškai. Pavyzdžiui, yra palikta kryžių aštuonakė, kryžių dešimtakė ir kryžių dama.

Trūkstamas kortas kryžių devynakę ir kryžių valetą parenkame vienareikšmiškai (140.6 pav.):

140.6 pav.

Atstumas tarp mažiausio ir didžiausio rangų turi būti mažesnis nei MAX_K. Kitaip negalėsime parinkti nei vienos kombinacijos. Pavyzdžiui, jei palikta kryžių devynakė ir kryžių tūzas, negalėsime sudėti nei vienos kombinacijos.

Sakykime, kad parinkus vidurines kortas vienareikšmiškai dar reikia parinkti kor kortų. Trūkstamas kortas galime prijungti arba prie kurio nors vieno, arba prie abiejų kortų eilutės galų. Pavyzdžiui, jei palikta kryžių valetas ir kryžių karalius, vienareikšmiškai parenkame kryžių damą. Likusias dvi kortas galime parinkti taip (140.7 pav.):

140.7 pav.

Iš vienos rūšies kortų galima sudaryti komb kombinacijų. Kiekviena nauja kombinacija gaunama keičiant kortų rūšis. Jei keičiant rūšį iš vienos kombinacijos galima gauti kmb kombinacijų, tai iš viso bus komb ( kmb kombinacijų. 

Rasime skaičių kmb. Imkime kurią nors kombinaciją, pavyzdžiui, kryžių devynakę, dešimtakę, valetą, damą, karalių. Keisime parinktų kortų kryžių devynakės, dešimtakės, damos rūšį. Keisdami devynakės rūšį, gausime keturias kombinacijas (140.8 pav.):

140.8 pav.

Keisdami dešimtakės rūšį iš kiekvienos turimos kombinacijos gausime dar po 4 kombinacijas. Iš viso turėsime 4(4 ( 16 kombinacijų. Keisdami damos rūšį vėl iš kiekvienos kombinacijos gausime po keturias naujas kombinacijas. Taigi, kombinacijų skaičius padidės iki 16(4 ( 64 kombinacijų. 

Gauname, kad kmb ( liko4, čia liko – parenkamų kortų skaičius. 

Bendras kombinacijų, sudarančių išlošiamą situaciją Iš eilės skaičius lygus 

komb ( liko4



Į šį skaičių įtrauktos ir išlošiamos kombinacijos Vienos spalvos bei Pokeris. Todėl radus jų skaičių, jį reikės atimti iš išlošiamos padėties Iš eilės kombinacijų skaičiaus. 

Rasime išlošiamos padėties Pokeris kombinacijų skaičių. Jei palikta nors viena korta, kombinacijų skaičius lygus anksčiau apskaičiuotam komb, mat spalvos keisti negalima. Jei nepalikta nei vienos kortos, jis lygus: komb ( 4. 

function kelk (a, n : longint): longint;

{ a kelia n-tuoju laipsniu }

  var i: longint;

      sk: longint;

begin

   sk := 1;

   for i := 1 to n do

      sk := sk * a;

   kelk := sk

end; { kelk }



function vienodos_spalvos (k: kortos; pal: log_mas): 

                                                 boolean;

{ patikrina, ar paliktos kortos yra vienodos spalvos }

  var sp: char;

      i: longint;

begin

  vienodos_spalvos := true;

  sp := ' ';       { nežinome pirmosios paliktos kortos spalvos }

  for i := 1 to 5 do  { patikrinama ar paliktos kortos yra }

    if pal[i] then    { vienodos spalvos }

       if sp = ' '  { jei tai pirma palikta korta }

          then sp := k[i].sp

          else if k[i].sp <> sp { yra nevienodos spalvos kortų }

                  then vienodos_spalvos := false

end; { vienodos_spalvos }



procedure iš_eilės_komb (liko: longint; pal: log_mas; 

                         k: kortos;

                         var kiek: išlosimas);

{ suskaičiuoja kiek yra galimų kombinacijų, kuriose kortos yra }

{ išdėliotos iš eiles; t. y. skaičiuoja keliais variantais sudaroma }

{ ketvirta ir aštunta išlošianti kombinacija }

  var komb, min, max, i, kr, kair, deš: longint;

begin

  { pradžioje skaičiuosime kombinacijas, tarsi turėtume tik }

  { vienos rūšies kortas }

  if liko = MAX_K then komb := MAX_R - MAX_K + 1

  else begin

        max := 1; min := MAX_R; { randamos mažiausio ir didžiausio }

        for i := 1 to MAX_K do { rango paliktos kortos }

          if pal[i]

            then begin

                  if k[i].ran > max then max := k[i].ran;

                  if k[i].ran < min then min := k[i].ran

                 end;

        if max - min >= MAX_K { jei atstumas tarp rangų per didelis }

           then komb := 0

           else begin

                  { tiek kortų reikia pridėti iš galų }

                  kr := MAX_K - (max - min + 1);

                  komb := 0;

                  for kair := 0 to kr do

                    begin

                      { tiek kortų pridėsime iš dešinio galo }

                      deš := kr - kair;

                      if (min - kair >= 1) and 

                         (max + deš <= MAX_R)

                          then komb := komb + 1;

                    end;

                end;

       end;

  { rasime kombinacijų skaičių, kai kortos yra keturių rūšių }

  kiek[4] := komb * kelk (4,liko); { kiekviena iš parenkamų kortų }

                                 { gali būti vienos iš keturių spalvų }

  if vienodos_spalvos (k, pal) { Pokeris }

     then begin

            if liko = 5 

               then kiek[8] := komb*4 { tinka bet kuri spalva }

               else kiek[8] := komb;  { tinka paliktų kortų spalva }

            { šios kombinacijos jau buvo priskaičiuotos prie ketvirtos }

            { išlošiamosios padėties }

             kiek[4] := kiek[4] - kiek[8]

          end

end; { iš_eilės_komb }



Beliko apskaičiuoti Vienos spalvos kombinacijų skaičių. Samprotaujama analogiškai, kaip ir skaičiuojant išlošiamų padėčių su keliomis vienodo rango kortomis (Pora, Dvi poros ir t. t.) kombinacijų skaičių. 

Visos parenkamos kortos bus skirtingo rango ir tos pačios spalvos. Reikia parinkti liko kortų. Jau yra parinkta MAX_K - liko kortų. Kadangi iš viso yra MAX_R skirtingo rango kortų, galima rinkti iš MAX_R - (MAX_K - liko) kortų. Vėl gauname Derinius be pasikartojimų, kur n ( MAX_R - (MAX_K - liko), o k ( liko.

procedure vien_spalvos_komb (liko: longint; 

                             var kiek: išlošimas);

{ randa kiek yra vienodos spalvos kombinacijų }

begin

  kiek[5] := cNpok (MAX_R-(5-liko), liko);

  if liko = 5 then kiek[5] := kiek[5]*4; { tinka bet kuri spalva }

  { atimamos Pokerio kombinacijos }

  kiek[5] :=  kiek[5] - kiek[8]

end; { vien_spalvos_komb }



Procedūras perrinkti, iš_eilės_komb ir vien_spalvos_komb sujungiame funkcija kokia_nauda.

function kokia_nauda (k: kortos; išl: išlošimas;

                      pal: log_mas { kiek atitinkamo rango kortų } 

                                   { yra palikta }: real;

{ skaičiuoja, koks išlosimo vidurkis pakeitus nurodytas kortas }

  var vien: array [1..4] of longint; { vien[i] > 0, jei yra i }

                                     { vienodų kortų }

      rng: rangai; { kiek tarp paliktų kortų yra šio rango kortų }

      kiek: išlošimas; { kelias kombinacijas galima sudaryti, kad }

                       { gautume i-tąją išlošiančiąją padėtį }

      išviso, { galimų skirtingų kombinacijų skaičius, keičiant }

              { nurodytas kortas }

      liko,  { kiek kortų reikia parinkti }

      vid, i: longint;

      skirt: boolean;

  procedure analizuok_paliktas(k: kortos; { duotosios kortos }

                               pal: log_mas; { paliktos kortos }

                               var rng: rangai;

                               var liko: longint);

  { suskaičiuoja kiek iš kortų reikia parinkti }

  { ir kiek kiekvieno rango kortų palikta }

    var i: longint;

  begin

    for i := 1 to  MAX_R do

      rng[i] := 0;

    liko := MAX_K;

    for i := 1 to MAX_K do

      if pal[i] then begin

                       liko := liko - 1;

                       rng[k[i].ran] := rng[k[i].ran] + 1

                     end;

  end; { analizuok_paliktas }

begin

  { suskaičiuojama kiek kortų reikia parinkti  ir kiek kokio rango kortų liko }

  analizuok_paliktas (k, pal, rng, liko);

  { kol kas nesudarėme nei vienos išlošiamos padėties }

  for i := 1 to MAX_P do

    kiek[i] := 0;

  for i := 1 to 4 do

    vien[i] := 0;

  išviso := 0; { nesudarėme nei vienos kombinacijos }

  { rasime visų galimų kombinacijų skaičių bei kombinacijų skaičių, kai }

  { susidaro išlošiančios padėtys: Pora, Dvi poros, Trejetas, Ketvertas }

  { į likusias kombinacijas nekreipiama dėmesio; procedūros rezultatas – }

  { masyvas kiek }

  perrinkti (1, liko, 1);

  { bandysime sudaryti likusias kombinacijas }

  { patikrinsime, ar paliktos kortos yra skirtingo rango }

  skirt := true;

  for i := 1 to MAX_R do

    skirt := skirt and (rng[i] <= 1);

  if skirt { visos kortos skirtingos }

     then begin { ieškosime 4-tos, 5-tos ir 8-tos išlošiančių }

            iš_eilės_komb (liko, pal, k, kiek);  { padėčių }

            if vienodos_spalvos (k, pal)

               then vien_spalvos_komb (liko, kiek)

          end;

  { apskaičiuojamas išlošimo vidurkis }

  vid := 0;

  for i := 1 to MAX_P do

    vid := vid + kiek[i] * išl[i];

  kokia_nauda := vid/išviso;

end; { kokia_nauda }



Testai yra suskirstyti poromis. Abiejuose vienos poros testuose pateiktos tos pačios kortos, tačiau skirtingos išlošimų lentelės. 

Testo nr.�Pateiktos kortos�Rezultatai�Maksimalus išlošimo vidurkis��1�V3 K3 S3 S2 S4�TTTNN� 77.143��2��NNTTT� 59.410��3�V3 V2 K6 S1 B5�TTNTT arba 

TTTFT� 10.600��4��TTNNN� 15.275��5�V2 K2 V3 K3 V4�TTTTN� 24.000��6��TNTNT� 30.243��7�K1 K2 K3 K4 K6�TTTTN arba

NTTTT�268.100��8��TTTTT�200.000��9�V1 V2 V3 K3 B4�NNTTN� 15.740��10��TTTNN� 15.005��Pastaba. Paliekamos kortos žymimos „T“ raide, keičiamos – „N“ raide.

142. Nežinomas kodavimas. Šio uždavinio sprendimas primena 133 uždavinio sprendimą. Todėl prieš pradėdami nagrinėti šį sprendimą išanalizuokite to uždavinio sprendimą.

Įvykdykime kod.exe su savo sukurta pradinių duomenų byla. Pasižiūrėję į rezultatų bylą pamatysime į kokio didumo grupes skirstomi bitai. Šiuo atveju bitų grupės ilgis yra 6.

Kaip ir ankstesnio uždavinio atveju, reikia išnagrinėti visas galimas šešių bitų grupes. Jų bus šešiasdešimt keturios (26): 000000, 000001, .... 111111. Susikuriame pradinių duomenų bylą, kurioje būtų visos įmanomos kombinacijos. 

Šį kartą bylą sukurti kiek sunkiau: keturios bitų grupės sujungtos kartu sudarys tris baitus. Bus geriausia, jei parašysime programą, kuriančią tokią bylą:

program kodavimo_sistemos_nustatymas;

{ programa sukuria bylą, kurioje eilės tvarka išdėstyti visi }

{ galimi šešių bitų grupės variantai: nuo 000000 iki 111111 }

  procedure bitų_grupės (var f: file);

    var kiek, num, i: integer;

        baitas1, baitas2, baitas3: byte;

        rezultatas: array [1..48] of byte;

  begin

    i := 0;   { bitų grupė }

    num := 1; { nagrinėjamo baito numeris }

    while i < 64 do

      begin

        { sudarysime keturias bitų grupes, kurios užims tris baitus }

        baitas1 := byte(i + (i + 1) shl 6);

        baitas2 := byte((i + 1) shl 2 shr 4 + (i + 2) shl 4);

        baitas3 := byte((i + 2) shr 4 + (i + 3) shl 2);

        rezultatas[num]   := baitas1;

        rezultatas[num+1] := baitas2;

        rezultatas[num+2] := baitas3;

        i := i + 4;

        num := num + 3;

      end;

    { 64 bitų grupės po 6 bitus sudarys 48 baitus }

    blockwrite (f, rezultatas, 48, kiek);

  end; { bitų_grupės }



  var f: file;

begin

  writeln ('Ši programa sukuria bylą TEST.DAT, 

            kurioje eilės tvarka');

  writeln ('surašytos 64 bitų grupės (po 6 bitus): 

            nuo 000000 iki 111111');

  assign (f, 'TEST.DAT');

  rewrite (f, 1);

  bitų_grupės (f);

  close (f);

end.



Sukūrę bylą test.dat ir ją užkodavę programa kod.exe, pamatysime, kad baitų grupės koduojamos tokia simbolių eilute:

0123456789ABCDEFGHIJKLMNOPQRSTUVWXYZabcdefghijklmnopqrstuvwxyz+-

T. y. bitų grupė „000000“ koduojama simboliu „0“, bitų grupė „000001“ koduojama simboliu „1“ ir t. t. Bitų grupė „111111“ koduojama simboliu „-“.

Žinodami kokiu būdu užkoduojama galime parašyti dekodavimo programą. Prieš tai panagrinėkime tokį pavyzdį. Sakykime, kad užkoduoti duomenys pavaizduoti tokia simbolių eilute: B96.

Paaiškinsime, kaip jie dekoduojami. Užkoduotą tekstą sudaro trys baitai. Randame visų trijų simbolių numerius kodavimui naudojamų simbolių eilutėje: simbolio „B“ numeris yra 12, simbolio „9“ – 10, ir simbolio „6“ – 7. 

Simbolio numeris, užrašytas dvejetaine sistema, turėtų sutapti su tuo simbolių užkoduota šešių bitų grupe. Tačiau pirmoji koduojama bitų grupė yra sudaryta vien iš nulių„000000“, o simboliai eilutėje numeruojami nuo 1. Taigi, norint rasti šešių bitų grupę, kuri buvo užkoduota, reikia iš simbolio numerio atimti vienetą. Todėl nagrinėsime skaičius 1110, 910, 610.

Šiuos skaičius užrašome dvejetaine sistema:

1110 ( 000010112;

 910 ( 000010012;

 610.( 000001102;

Žinome, kad fiziškai kompiuteryje baitai laikomi nuo jauniausio iki vyriausio. Imame po šešis jaunesnius kiekvieno baito bitus ir juos surašome vienas po kito. Rašome nuo jauniausio iki vyriausio:

110100100100011000

Atskiriame pirmuosius aštuonis bitus (11010010) juos perrašome nuo vyriausio iki jauniausio: 010010112 ( 4B16 ( 7510. ASCII kodų lentelėje tai yra simbolio „K“ kodas. Imame antrąjį baitą: 011000102 ( 6216 ( 9810. ASCII kodų lentelėje tai yra simbolio „b“ kodas. Likę du nuliniai bitai „00“ baito nebesudaro, taigi į juos dėmesio nebekreipiame. Kadangi užkoduojamuose duomenyse bitų skaičius nesidalijo iš 6, teko prirašyti du papildomus nulinius bitus.

Atkodavę duomenis, gavome „Kb“.

Pateikiame programos kod.exe pradinį tekstą:

program kodavimas_6_bitais;



  const

   BITAI = 6;         { bitų grupės ilgis }

   LENT: string[64] = { kodavimui naudojami simboliai }

'0123456789ABCDEFGHIJKLMNOPQRSTUVWXYZabcdefghijklmnopqrstuvwxyz+-';

   MAX = 10000; { maksimalus rezultatų bylos dydis baitais }



  procedure klaida;

  { nutraukia programos darbą dėl įvedimo/išvedimo klaidos }

  begin

    writeln ('Įvedimo/išvedimo klaida!');

    halt;

  end; { klaida }



  procedure atidar_bylas (var fskait, fraš: file);

  { patikrina parametrus ir atidaro bylas }

  begin

    writeln;

    if paramcount <> 2 { jei nenurodyti tinkami parametrai }

       then begin

              writeln ('Naudojimas: ');

              writeln( ' KOD.EXE  duom.dat rez.dat' );

              writeln;

              halt;

            end;

    assign (fskait, paramstr(1));

    {$I-}

    reset (fskait, 1);

    {$I+}

    if ioresult <> 0

       then klaida;

    assign (fraš, paramstr(2));

    {$I-}

    rewrite (fraš, 1);

    {$i+}

    if ioresult <> 0

       then klaida;

  end; { atidar_bylas }



  procedure uždar_bylas (var fskait, fraš: file);

  begin

    close (fraš);

    close (fskait);

  end; { uždar_bylas }



  procedure kodavimas (var fskait, fraš: file);

  { užkoduoja duomenis }

    var duom, rez: array [1..MAX] of byte;

        dydis, { pradinių duomenų bylos dydis baitais }

        rsk,   { rezultatų bylos dydis baitais }

        i, { nagrinėjamo baito numeris }

        perskait, poslinkis, kaukė: integer;

        bit6: byte; { šešių bitų grupė }

  begin

    { inicializuojame pradinių duomenų ir rezultatų masyvus }

    fillchar (duom, sizeof(duom), 0);

    fillchar (rez, sizeof(rez), 0);

    { randame pradinių duomenų kiekį baitais }

    dydis := filesize (fskait);

    { perskaitome pradinius duomenis }

    blockread (fskait, duom, dydis, perskait);

    if perskait <> dydis

       then begin

              uždar_bylas (fskait, fraš);

              klaida;

            end;

    i := 1;

    poslinkis := 0;

    rsk := 0;

    kaukė := $3F; { 00111111 }

    while i <= dydis do

      begin

        rsk := rsk + 1;

        { paimame eilinius 6 bitus }

        if i = dydis

           then bit6 := (duom[i] shr poslinkis) and kaukė

           else bit6 := (duom[i] shr poslinkis + 

                         duom[i+1] shl (8-poslinkis))

                         and kaukė;

        { juos užkoduojame vienu baitu kurį įrašome į rezultatų masyvą }

        rez[rsk] := ord (lent[bit6 + 1]);

        poslinkis := poslinkis + BITAI;

        if poslinkis > 7

           then begin

                  poslinkis := poslinkis - 8;

                  i := i + 1

                end;

      end;

    { įrašome užkoduotus duomenis į rezultatų bylą }

    blockwrite (fraš, rez, rsk, dydis);

    if rsk <> dydis

       then klaida;

  end; { kodavimas }



  var fskait, fraš: file;

begin

  atidar_bylas (fskait, fraš);

  kodavimas (fskait, fraš);

  uždar_bylas (fskait, fraš);

end.

Žemiau pateikiame procedūrą, atkoduojančią duomenis. Pagrindinė programos dalis bei procedūra klaida (į ją kreipiasi procedūra dekodavimas) yra analogiškos aukščiau pateiktoje programoje esančiai pagrindinei programos daliai bei procedūrai klaida.

const BITAI = 6;         { bitų grupės ilgis }

      MAX = 10000; { maksimalus užkoduotos bylos dydis baitais }

      LENT: string[64] = { kodavimui naudojami simboliai }

'0123456789ABCDEFGHIJKLMNOPQRSTUVWXYZabcdefghijklmnopqrstuvwxyz+-';



procedure dekodavimas (var fskait: text; var fraš: file);

{ dekoduoja duomenis }

  var rez: array [1..MAX] of byte;

      num, nr, poslinkis, dydis: integer;

      simb: char;



  function numeris (c: char): integer;

  { randa simbolio c numerį lentelėje LENT }

    var rasta: boolean;

        i: integer;

  begin

    i := 0; rasta := false;

    while (i < 64) and not rasta do

      begin

        i := i + 1;

        rasta := c = LENT[i];

      end;

    if rasta

       then numeris := i

       else numeris := 0

  end; { numeris }



begin

  { inicializuojame rezultatų masyvą }

  fillchar (rez, sizeof(rez), 0);

  poslinkis := 0; { poslinkis baito pradžios atžvilgiu }

  nr := 1; { baito numeris atkoduotame tekste }

  while not eof (fskait) do

    begin

      { iš užkoduotos bylos perskaitome simbolį }

      read (fskait, simb);

      { randame jo numerį simbolių eilutėje lent }

      num := numeris (simb);

      { jei tokio numerio nerasta }

      if num = 0

         then klaida;

      { dekoduojame baitą }

      rez[nr] := byte(rez[nr] + (num-1) shl poslinkis);

      if poslinkis > 2

         then rez[nr+1] := byte((num-1) shr (8 - poslinkis));

      { perskaičiuojame poslinkį }

      poslinkis := poslinkis + BITAI;

      if poslinkis > 7

         then begin

                poslinkis := poslinkis - 8;

                nr := nr + 1

              end;

    end;

  { sumažiname nr reikšmę: arba ji buvo padidinta be reikalo }

  { arba paskutinysis baitas užpildytas užkoduojant papildomai }

  { prirašytais nuliais }

  nr := nr - 1;

  { įrašome dekoduotus duomenis į rezultatų bylą }

  blockwrite (fraš, rez, nr, dydis);

  if nr <> dydis

     then klaida;

end; { kodavimas }



Testo nr.�Paaiškinimai�Užkoduoti duomenys�Atkoduoti duomenys��1�Užkoduotą testą sudaro ta pati 64 simbolių eilutė, kuria koduojamos šešių bitų grupės�0123456789 ir t. t.���2�Užkoduoto testo ilgis – 1 baitas (pabaigoje prirašyti 4 nuliai). �b0�%��3�Užkoduoto testo ilgis – 2 baitai (pabaigoje prirašyti 2 nuliai)�B96�Kb��4�Užkoduoto testo ilgis – 3 baitai (pabaigoje prirašytų nulių nėra). Atkoduotą tekstą sudaro simbolis „5“ ir eilutės pabaigos simbolis�rqW2�5��5�Teste surašyti visi simboliai, kurių kodai yra nuo 0016 iki FF16�04W03GG16S029 ir t. t.���

Pastaba. Atsakymuose ne visur pateikėme atkoduotus duomenis: jų tarpe yra valdančių bei pseudografikos simbolių, taip pat simbolių, kurių kodai priklauso nuo konkrečios kodų lentelės.



143. Lošimas disku. Iš pažiūros šis uždavinys nesiskiria nuo kitų lošimo uždavinių, kurie sprendžiami rekursiškai bandant visus ėjimus ir einant „žemyn“ nuo duotosios situacijos link žaidimo pabaigos situacijos. Tačiau sąlygoje prašoma ne tik rasti ėjimą, bet ir paaiškinti žaidimo strategiją. O rekursijos žaidimo strategija nepavadinsi. Be to, vykdymo laikas skirtas ne vienam testui, o 12000 testų. Taigi, šis būdas atpuola.

Norint aprašyti žaidimo strategiją, teks panagrinėti laimimas ir pralaimimas situacijas. Žinome, sukauptą sumą s ir sumą, kurią reikia pasiekti m. Mus domina tik jų skirtumas liko ( s - m, t. y. suma, kurią dar reikia sukaupti. 

Susidarysime laiminčių situacijų lentelę. liko reikšmę didinsime  nuo 1 iki ims aiškėti žaidimo strategija. Lentelę sudarys dvi skiltys. Vienoje jų bus liko reikšmė, kitoje bus surašyti visi skaičiai, kuriems atsiradus viršutiniame disko segmente po disko pasukimo galima laimėti. 

Sakykime, kad liko ( 1. Laimėti galima tik tada, kai pasukus diską viršutiniame disko segmente yra 1. Likusiais atvejais žaidėjas pralaimi. 

Jei liko ( 2, žaidėjas laimi, kai atsisuka 2 arba 1. Jei jis atsisuka 2, tuomet pasiekia sutartą skaičių. Jei atsisuka 1, tuomet liko reikšmė sumažėja iki 1 ir disko viršuje lieka 1. Priešininkas laimėti nebegalės, nes diską jis privalės pasukti, o pasukęs niekaip negalės disko viršuje gauti 1 – skaičiaus, kuris turi atsidurti disko viršutiniame segmente, norint laimėti, kai liko reikšmė lygi 1.

Taip žiūrėdami į jau išnagrinėtas situacijas, pildome lentelę:

liko reikšmė�skaičius atsiradęs viršutiniame disko segmente, pasukus diską�liko reikšmė�skaičius atsiradęs viršutiniame disko segmente, pasukus diską��1�1�15�5, 3, 1��2�2, 1�16�6, 5, 1��3�3�17�5, 3��4�4, 3, 1�18�3, 1��5�5, 4, 1�19�5, 3, 1��6�6, 5, 4, 3, 1�20�5, 3, 1��7�5�21�3, 1��8�5, 1�22�5, 3, 1��9�6, 2, 1�23�5, 3, 1��10�3, �24�5, 3, 1��11�3, 1�25�5, 3, 1��12�5, 1�26�5, 3, 1��13�5, 4, 3, 1�27�5, 3, 1��14�3, 1�28, 29, ...�5, 3, 1��

Matome, kad norint laimėti, kai liko reikšmės yra didesnės už 21, pasukus diską turi atsisukti arba 1, arba 3, arba 5. Norėdami tai gauti, diską turime sukti į .... bet kurią pusę... Mat jei ėjimo pradžioje disko viršuje buvo vienas iš skaičių 1, 3, 5, į kurią pusę besuktume diską, disko viršuje atsiras 2, 4, arba 6. Iš kitos pusės jei ėjimo pradžioje disko viršuje buvo 2, 4 ar 6, pasukus diską į kairę ar į dešinę, atsisuks 1, 3 ar 5.

Beliko rasti tinkamus ėjimus mažesnėms liko reikšmėms ir juos surašyti į funkciją diskas.

function diskas (n, s, m: integer): integer;

                { n – skaičius disko viršuje }

                { s – sukaupta suma }

                { m – galutinė suma }

  var krit: array [1..6] of set of 1..21;

          { kritinės situacijos, apsprendžiančios laimėjimą  }

          { norint laimėti prie šių situacijų, būtina diską sukti į kairę }

      ka, { koks skaičius bus viršuje diską pasukus į kairę  }

      de, { koks skaičius bus viršuje diską pasukus į dešinę }

      liko: integer; { kiek dar liko iki sutarto skaičiaus  }

begin

  { apskaičiuojame kritines situacijas }

  krit[1] := [2];

  krit[2] := [3, 10, 17];

  krit[3] := [4, 5, 13];

  krit[4] := [5, 7, 8, 12, 16];

  krit[5] := [9, 16];

  krit[6] := [1, 2, 4, 9, 11, 14, 18, 21];

  { rasime skaičius, kurie atsiras disko viršuje pasukus diską }

  if n = 6 then ka := 1

           else ka := n + 1;

  if n = 1 then de := 6

           else de := n - 1;

  liko := m-s;

  if liko > 21

    then diskas := ka

    else if liko in krit[n]

      then diskas := ka

      else diskas := de

end; { diskas }



Sprendimas buvo tikrinamas su 12000 pradinių duomenų variantų: m ( 2000, s kito nuo 0 iki m-1, n – nuo 1 iki 6.

Žemiau pateikti tik šeši atsitiktinai parinkti testai:

Testo nr.�Pradiniai duomenys�Galimi rezultatai�Paaiškinimai��1�n ( 6, s ( 2, m ( 20;�1�Sukama į kairę��2�n ( 4, s ( 4, m ( 25;�3�Sukama į dešinę��3�n ( 1, s ( 14, m ( 20;�6�Sukama į dešinę��4�n ( 13, s ( 17, m ( 20;�4�Sukama į kairę��5�n ( 4, s ( 3, m ( 20;�3 5�Laimima bet kuriuo atveju��6�n ( 3, s ( 12, m ( 20;�4 2�Pralaimima bet kuriuo atveju��

